
Figure 6.33: The Process Browser

6.7 Finding methods

There are two tools in Pharo to help you find messages. They differ in both
interface and functionality.

The method finder was described at some length in Section 1.9; you can
use it to find methods by name or by functionality. However, to look at the
body of a method, the method finder opens a new browser. This can quickly
become overwhelming.

The message names browser has more limited search functionality: you
type a fragment of a message selector in the search box, and the browser lists
all methods that contain that fragment in their names, as shown in Figure 6.34.
However, it is a full-fledged browser: if you select one of the names in the
left pane, all of the methods with that name are listed in the right pane, and
can be browsed in the bottom pane. As with the browser, the message names
browser has a button bar that can be used to open other browsers on the
selected method or its class.

6.8 Change sets and the Change Sorter

Whenever you are working in Pharo, any changes that you make to meth-
ods and classes are recorded in a change set. This includes creating new
classes, re-naming classes, changing categories, adding methods to existing
classes — just about everything of significance. However, arbitrary doits are

Figure 6.34: The message names browser showing all methods containing the
substring random in their selectors.

not included, so if, for example, you create a new global variable by assigning
to it in a workspace, the variable creation will not make it into a change set.

At any time, many change sets exist, but only one of them — ChangeSet
current — is collecting the changes that are being made to the image. You can
see which change set is current and can examine all of the change sets using
the change sorter, available by selecting World .Tools . . . .Change Sorter .

Figure 6.35 shows this browser. The title bar shows which change set is
current, and this change set is selected when the change sorter opens.

Other change sets can be selected in the top-left pane; the action-click
menu allows you to make a different change set current, or to create a new
change set. The next pane lists all of the classes affected by the selected
change set (with their categories). Selecting one of the classes displays the
names of those of its methods that are also in the change set (not all of the
methods in the class) in the left central pane, and selecting a method name
displays the method definition in the bottom pane. Note that the change
sorter does not show you whether the creation of the class itself is part of the
change set, although this information is stored in the object structure that is

Figure 6.35: The Change Sorter

used to represent the change set.

The change sorter also lets you delete classes and methods from the change
set using the action-click menu on the corresponding items.

The change sorter allows you to simultaneously view two change sets,
one on the left hand side and the other on the right. This layout supports the
change sorter’s main feature, which is the ability to move or copy changes
from one change set to another, as shown by the action-click menu in Fig-
ure 6.35. It is also possible to copy individual methods from one side to the
other.

You may be wondering why you should care about the composition of a
change set. the answer is that change sets provide a simple mechanism for
exporting code from Pharo to the file system, from where it can be imported
into another Pharo image, or into another non-Pharo Smalltalk. Change set
export is known as “filing-out”, and can be accomplished using the action-
click menu on any change set, class or method in either browser. Repeated
file outs create new versions of the file, but change sets are not a versioning
tool like Monticello: they do not keep track of dependencies.

Before the advent of Monticello, change sets were the main means for
exchanging code between Pharoers. They have the advantage of simplicity
(the file out is just a text file, although we don’t recommend that you try to
edit them with a text editor), and a degree of portability.

The main drawback of change sets, compared to Monticello packages,
is that they do not support the notion of dependencies. A filed-out change

set is a set of actions that change any image into which it is loaded. To
successfully load a change set requires that the image be in an appropriate
state. For example, the change set might contain an action to add a method
to a class; this can only be accomplished if the class is already defined in the
image. Similarly, the change set might rename or re-categorize a class, which
obviously will only work if the class is present in the image; methods may
use instance variables that were declared when they were filed out, but which
do not exist in the image into which they are imported. The problem is that
change sets do not explicitly describe the conditions under which they can
be filed in: the file in process just hopes for the best, usually resulting in a
cryptic error message and a stack trace when things go wrong. Even if the
file in works, one change set might silently undo a change made by another
change set.

In contrast, Monticello packages represent code in a declarative fashion:
they describe the state of the image should be in after they have been loaded.
This permits Monticello to warn you about conflicts (when two packages
require contradictory final states) and to offer to load a series of packages in
dependency order.

In spite of these shortcomings, change sets still have their uses; in partic-
ular, you may find change sets on the Internet that you want to look at and
perhaps use. So, having filed out a change set using the change sorter, we
will now tell you how to file one in. This requires the use of another tool, the
file list browser.

6.9 The File List Browser

The file list browser is in fact a general-purpose tool for browsing the
file system (and also FTP servers) from Pharo. You can open it from the
World.Tools . . . .File Browser menu. What you see of course depends on the
contents of your local file system, but a typical view is shown in Figure 6.36.

When you first open a file list browser it will be focussed on the current
directory, that is, the one from which you started Pharo. The title bar shows
the path to this directory. The larger pane on the left-hand side can be used
to navigate the file system in the conventional way. When a directory is
selected, the files that it contains (but not the directories) are displayed on
the right. This list of files can be filtered by entering a Unix-style pattern in
the small box at the top-left of the window. Initially, this pattern is *, which
matches all file names, but you can type a different string there and accept it,
changing the pattern. (Note that a * is implicitly prepended and appended to
the pattern that you type.) The sort order of the files can be changes using the
name , date and size buttons. The rest of the buttons depend on the name
of the file selected in the browser. In Figure 6.36, the file name has the suffix

Figure 6.36: A file list browser

.cs, so the browser assumes that it is a change set, and provides buttons to
install it (which files it in to a new change set whose name is derived from the
name of the file), to browse the changes in the file, to examine the code in the
file, and to filein the code into the current change set. You might think that the
conflicts button would tell you about changes in the change set that conflicted
with existing code in the image, but it doesn’t. Instead it just checks for
potential problems in the file that might indicate that the file cannot properly
be loaded (such as the presence of linefeeds).

Because the choice of buttons to display depends on the file’s name, and
not on its contents, sometimes the button that you want won’t be on the
screen. However, the full set of options is always available from the action-
click more . . . menu, so you can easily work around this problem.

The code button is perhaps the most useful for working with change sets;
it opens a browser on the contents of the change set file; an example is shown
in Figure 6.37. The file contents browser is similar to the browser except that it
does not show categories, just classes, protocols and methods. For each class,
the browser will tell you whether the class already exists in the system and
whether it is defined in the file (but not whether the definitions are identical).
It will show the methods in each class, and (as shown in Figure 6.37) will
show you the differences between the current version and the version in the
file. Contextual menu items in each of the top four panes will also let you
file in the whole of the change set, or the corresponding class, protocol or
method.



In Smalltalk, you can’t lose code 143

Figure 6.37: A File Contents Browser

6.10 In Smalltalk, you can’t lose code

It is quite possible to crash Pharo: as an experimental system, Pharo lets you
change anything, including things that are vital to make Pharo work!

To maliciously crash Pharo, try Object become: nil.

The good news is that you need never lose any work, even if you crash
and go back to the last saved version of your image, which might be hours
old. This is because all of the code that you executed is saved in the .changes
file. All of it! This includes one liners that you evaluate in a workspace, as
well as code that you add to a class while programming.

So here are the instructions on how to get your code back. There is no
need to read this until you need it. However, when you do need it, you’ll find
it here waiting for you.

In the worst case, you can use a text editor on the .changes file, but since it
is many megabytes in size, this can be slow and is not recommended. Pharo
offers you better ways.

How to get your code back

Restart Pharo from the most recent snapshot, and select
World.Tools . . . .Recover lost changes .

This will give the opportunity to decide how far back in history you

wish to browse. Normally, it’s sufficient to browse changes as far back as
the last snapshot. (You can get much the same effect by editing ChangeList
browseRecent: 2000 so that the number 2000 becomes something else, using
trial and error.)

One you have a recent changes browser, showing, say, changes back as
far as your last snapshot, you will have a list of everything that you have
done to Pharo during that time. You can delete items from this list using the
action-click menu. When you are satisfied, you can file-in what is left, thus
incorporating the changes into your new image. It’s a good idea to start a
new change set, using the ordinary change set browser, before you do the file
in, so that all of your recovered code will be in a new change set. You can
then file out this change set.

One useful thing to do in the recent changes browser is to remove doIts .
Usually, you won’t want to file in (and thus re-execute) doIts. However, there
is an exception. Creating a class shows up as a doIt . Before you can file in the
methods for a class, the class must exist. So, if you have created any new classes,
first file-in the class creation doIts, then remove doIts and file in the methods.

When I am finished with the recovery, I like to file out my new change
set, quit Pharo without saving the image, restart, and make sure that the new
change set files back in cleanly.

6.11 Chapter summary

In order to develop effectively with Pharo, it is important to invest some effort
into learning the tools available in the environment.

• The standard browser is your main interface for browsing existing cate-
gories, classes, method protocols and methods, and for defining new
ones. The browser offers several useful buttons to directly jump to
senders or implementors of a message, versions of a method, and so on.

• There exist several different browsers (such as the OmniBrowser and
the Refactoring Browser), and several specialized browsers (such as
the hierarchy browser) which provide different views of classes and
methods.

• From any of the tools, you can highlight the name of a class or a method
and immediately jump to a browser by using the keyboard shortcut
CMD–b.

• You can also browse the Smalltalk system programmatically by sending
messages to SystemNavigation default.

• Monticello is a tool for exporting, importing, versioning and sharing
packages of classes and methods. A Monticello package consists of a
category, subcategories, and related methods protocols in other cate-
gories.

• The inspector and the explorer are two tools that are useful for exploring
and interacting with live objects in your image. You can even inspect
tools by meta-clicking to bring up their morphic halo and selecting the
debug handle .

• The debugger is a tool that not only lets you inspect the run-time stack of
your program when an error is raised, but it also enables you to interact
with all of the objects of your application, including the source code. In
many cases you can modify your source code from the debugger and
continue executing. The debugger is especially effective as a tool to
support test-first development in tandem with SUnit (Chapter 7).

• The process browser lets you monitor, query and interact with the pro-
cesses current running in your image.

• The method finder and the message names browser are two tools for locating
methods. The first is more useful when you are not sure of the name, but
you know the expected behaviour. The second offers a more advanced
browsing interface when you know at least a fragment of the name.

• Change sets are automatically generated logs of all changes to the source
code of your image. They have largely been superseded by Monticello
as a means to store and exchange versions of your source code, but are
still useful, especially for recovering from catastrophic failures, however
rare these may be.

• The file list browser is a tool for browsing the file system. It also allows
you to filein source code from the file system.

• In case your image crashes before you could save it or backup your
source code with Monticello, you can always recover your most recent
changes using a change list browser. You can then select the changes you
want to replay and file them into the most recent copy of your image.





Chapter 7

SUnit

7.1 Introduction

SUnit is a minimal yet powerful framework that supports the creation and
deployment of tests. As might be guessed from its name, the design of SUnit
focussed on Unit Tests, but in fact it can be used for integration tests and
functional tests as well. SUnit was originally developed by Kent Beck and
subsequently extended by Joseph Pelrine and others to incorporate the notion
of a resource, which we will describe in Section 7.6.

The interest in testing and Test Driven Development is not limited to
Pharo or Smalltalk. Automated testing has become a hallmark of the Agile
software development movement, and any software developer concerned
with improving software quality would do well to adopt it. Indeed, devel-
opers in many languages have come to appreciate the power of unit testing,
and versions of xUnit now exist for many languages, including Java, Python,
Perl, .Net and Oracle. This chapter describes SUnit 3.3 (the current version
as of this writing); the official web site of SUnit is sunit.sourceforge.net, where
updates can be found.

Neither testing, nor the building of test suites, is new: everybody knows
that tests are a good way to catch errors. eXtreme Programming, by making
testing a core practice and by emphasizing automated tests, has helped to make
testing productive and fun, rather than a chore that programmers dislike.
The Smalltalk community has a long tradition of testing because of the incre-
mental style of development supported by its programming environment.
In traditional Smalltalk development, the programmer would write tests in
a workspace as soon as a method was finished. Sometimes a test would be
incorporated as a comment at the head of the method that it exercised, or
tests that needed some set up would be included as example methods in
the class. The problem with these practices is that tests in a workspace are

sunit.sourceforge.net

not available to other programmers who modify the code; comments and
example methods are better in this respect, but there is still no easy way to
keep track of them and to run them automatically. Tests that are not run do
not help you to find bugs! Moreover, an example method does not inform
the reader of the expected result: you can run the example and see the —
perhaps surprising — result, but you will not know if the observed behaviour
is correct.

SUnit is valuable because it allows us to write tests that are self-checking:
the test itself defines what the correct result should be. It also helps us to
organize tests into groups, to describe the context in which the tests must
run, and to run a group of tests automatically. In less than two minutes you
can write tests using SUnit, so instead of writing small code snippets in a
workspace, we encourage you to use SUnit and get all the advantages of
stored and automatically executable tests.

In this chapter we start by discussing why we test, and what makes a good
test. We then present a series of small examples showing how to use SUnit.
Finally, we look at the implementation of SUnit, so that you can understand
how Smalltalk uses the power of reflection in supporting its tools.

7.2 Why testing is important

Unfortunately, many developers believe that tests are a waste of their time.
After all, they do not write bugs — only other programmers do that. Most of
us have said, at some time or other: “I would write tests if I had more time.”
If you never write a bug, and if your code will never be changed in the future,
then indeed tests are a waste of your time. However, this most likely also
means that your application is trivial, or that it is not used by you or anyone
else. Think of tests as an investment for the future: having a suite of tests will
be quite useful now, but it will be extremely useful when your application, or
the environment in which it executes, changes in the future.

Tests play several roles. First, they provide documentation of the func-
tionality that they cover. Moreover, the documentation is active: watching
the tests pass tells you that the documentation is up-to-date. Second, tests
help developers to confirm that some changes that they have just made to
a package have not broken anything else in the system — and to find the
parts that break when that confidence turns out to be misplaced. Finally,
writing tests at the same time as — or even before — programming forces you
to think about the functionality that you want to design, and how it should
appear to the client, rather than about how to implement it. By writing the tests
first — before the code — you are compelled to state the context in which your
functionality will run, the way it will interact with the client code, and the
expected results. Your code will improve: try it.

We cannot test all aspects of any realistic application. Covering a complete
application is simply impossible and should not be the goal of testing. Even
with a good test suite some bugs will still creep into the application, where
they can lay dormant waiting for an opportunity to damage your system. If
you find that this has happened, take advantage of it! As soon as you uncover
the bug, write a test that exposes it, run the test, and watch it fail. Now you
can start to fix the bug: the test will tell you when you are done.

7.3 What makes a good test?

Writing good tests is a skill that can be learned most easily by practicing. Let
us look at the properties that tests should have to get a maximum benefit.

1. Tests should be repeatable. You should be able to run a test as often as
you want, and always get the same answer.

2. Tests should run without human intervention. You should even be able
to run them during the night.

3. Tests should tell a story. Each test should cover one aspect of a piece of
code. A test should act as a scenario that you or someone else can read
to understand a piece of functionality.

4. Tests should have a change frequency lower than that of the functional-
ity they cover: you do not want to have to change all your tests every
time you modify your application. One way to achieve this is to write
tests based on the public interfaces of the class that you are testing.
It is OK to write a test for a private “helper” method if you feel that
the method is complicated enough to need the test, but you should be
aware that such a test may have to be changed, or thrown away entirely,
when you think of a better implementation.

A consequence of property (3) is that the number of tests should be some-
what proportional to the number of functions to be tested: changing one
aspect of the system should not break all the tests but only a limited number.
This is important because having 100 tests fail should send a much stronger
message than having 10 tests fail. However, it is not always possible to
achieve this ideal: in particular, if a change breaks the initialization of an
object, or the set-up of a test, it is likely to cause all of the tests to fail.

eXtreme Programming advocates writing tests before writing code. This
may seem to go against our deep instincts as software developers. All we can
say is: go ahead and try it. We have found that writing the tests before the
code helps us to know what we want to code, helps us know when we are
done, and helps us conceptualize the functionality of a class and to design its

interface. Moreover, test-first development gives us the courage to go fast,
because we are not afraid that we will forget something important.

7.4 SUnit by example

Before going into the details of SUnit, we will show a step by step example.
We use an example that tests the class Set. Try entering the code as we go
along.

Step 1: create the test class

First you should create a new subclass of TestCase called ExampleSetTest. Add
two instance variables so that your new class looks like this:

Class 7.1: An Example Set Test class
TestCase subclass: #ExampleSetTest

instanceVariableNames: 'full empty'
classVariableNames: ''
poolDictionaries: ''
category: 'MySetTest'

We will use the class ExampleSetTest to group all the tests related to the
class Set. It defines the context in which the tests will run. Here the context
is described by the two instance variables full and empty that we will use to
represent a full and an empty set.

The name of the class is not critical, but by convention it should end
in Test. If you define a class called Pattern and call the corresponding test
class PatternTest, the two classes will be alphabetized together in the browser
(assuming that they are in the same category). It is critical that your class be a
subclass of TestCase.

Step 2: initialize the test context

The method setUp defines the context in which the tests will run, a bit like an
initialize method. setUp is invoked before the execution of each test method
defined in the test class.

Define the setUp method as follows, to initialize the empty variable to refer to an
empty set and the full variable to refer to a set containing two elements.

Method 7.2: Setting up a fixture
ExampleSetTest»setUp

empty := Set new.
full := Set with: 5 with: 6

In testing jargon the context is called the fixture for the test.

Step 3: write some test methods

Let’s create some tests by defining some methods in the class ExampleSetTest.
Each method represents one test; the name of the method should start with
the string ‘test’ so that SUnit will collect them into test suites. Test methods
take no arguments.

Define the following test methods.

The first test, named testIncludes, tests the includes: method of Set. The test
says that sending the message includes: 5 to a set containing 5 should return
true. Clearly, this test relies on the fact that the setUp method has already run.

Method 7.3: Testing set membership
ExampleSetTest»testIncludes

self assert: (full includes: 5).
self assert: (full includes: 6)

The second test, named testOccurrences, verifies that the number of occur-
rences of 5 in full set is equal to one, even if we add another element 5 to the
set.

Method 7.4: Testing occurrences
ExampleSetTest»testOccurrences

self assert: (empty occurrencesOf: 0) = 0.
self assert: (full occurrencesOf: 5) = 1.
full add: 5.
self assert: (full occurrencesOf: 5) = 1

Finally, we test that the set no longer contains the element 5 after we have
removed it.

Method 7.5: Testing removal
ExampleSetTest»testRemove

full remove: 5.
self assert: (full includes: 6).
self deny: (full includes: 5)

Note the use of the method deny: to assert something that should not be
true. aTest deny: anExpression is equivalent to aTest assert: anExpression not, but
is much more readable.

Step 4: run the tests

The easiest way to run the tests is directly from the browser. Simply action-
click on the package, class name, or on an individual test method, and select
run the tests (t) . The test methods will be flagged red or green, depending on
whether they pass or not, and the class will be flagged fully or partially green
or red depending on whether all, some or none of the tests pass.

Figure 7.1: Running SUnit tests from the browser

You can also select sets of test suites to run, and obtain a more detailed log
of the results using the SUnit Test Runner, which you can open by selecting
World .Test Runner . The TestRunner, shown in Figure 7.2, is designed to make
it easy to execute groups of tests. The left-most pane lists all of the categories
that contain test classes (i.e., subclasses of TestCase); when some of these
categories are selected, the test classes that they contain appear in the pane
to the right. Abstract classes are italicized, and the test class hierarchy is
shown by indentation, so subclasses of ClassTestCase are indented more than
subclasses of TestCase.

Open a Test Runner, select the category MyTest , and click the Run Selected
button.

Figure 7.2: The Pharo SUnit Test Runner

Introduce a bug in ExampleSetTest»testRemove and run the tests again. For
example, change 5 to 4.

The tests that did not pass (if any) are listed in the right-hand panes of the
Test Runner; if you want to debug one, to see why it failed, just click on the
name.

Step 5: interpret the results

The method assert: , which is defined in the class TestCase, expects a boolean
argument, usually the value of a tested expression. When the argument is
true, the test passes; when the argument is false, the test fails.

There are actually three possible outcomes of a test. The outcome that we
hope for is that all of the assertions in the test are true, in which case the test
passes. In the test runner, when all of the tests pass, the bar at the top turns
green. However, there are also two kinds of thing that can go wrong when
you run a test. Most obviously, one of the assertions can be false, causing the
test to fail. However, it is also possible that some kind of error occurs during
the execution of the test, such as a message not understood error or an index out

of bounds error. If an error occurs, the assertions in the test method may not
have been executed at all, so we can’t say that the test has failed; nevertheless,
something is clearly wrong! In the test runner, failing tests cause the bar at
the top to turn yellow, and are listed in the middle pane on the right, whereas
erroneous tests cause the bar to turn red, and are listed in the bottom pane on
the right.

Modify your tests to provoke both errors and failures.

7.5 The SUnit cook book

This section will give you more details on how to use SUnit. If you have used
another testing framework such as JUnit1, much of this will be familiar, since
all these frameworks have their roots in SUnit. Normally you will use SUnit’s
GUI to run tests, but there are situations where you may not want to use it.

Other assertions

In addition to assert: and deny:, there are several other methods that can be
used to make assertions.

First, assert:description: and deny:description: take a second argument which
is a message string that can be used to describe the reason for the failure, if it
is not obvious from the test itself. These methods are described in Section 7.7.

Next, SUnit provides two additional methods, should:raise: and
shouldnt:raise: for testing exception propagation. For example, you would use
(self should: aBlock raise: anException) to test that a particular exception is raised
during the execution of aBlock. Method 7.6 illustrates the use of should:raise:.

Try running this test.

Note that the first argument of the should: and shouldnt: methods is a block
that contains the expression to be evaluated.

Method 7.6: Testing error raising
ExampleSetTest»testIllegal

self should: [empty at: 5] raise: Error.
self should: [empty at: 5 put: #zork] raise: Error

SUnit is portable: it can be used from all dialects of Smalltalk. To make
SUnit portable, its developers factored-out the dialect-dependent aspects.
The class method TestResult class»error answers the system’s error class in a

1http://junit.org

http://junit.org

dialect-independent fashion. You can take advantage of this: if you want to
write tests that will work in any dialect of Smalltalk, instead of method 7.6
you would write:

Method 7.7: Portable error handling
ExampleSetTest»testIllegal

self should: [empty at: 5] raise: TestResult error.
self should: [empty at: 5 put: #zork] raise: TestResult error

Give it a try.

Running a single test

Normally, you will run your tests using the Test Runner. If you don’t want to
launch the Test Runner from the open . . . menu, you can execute TestRunner
open as a print it .

You can run a single test as follows.

ExampleSetTest run: #testRemove −→ 1 run, 1 passed, 0 failed, 0 errors

Running all the tests in a test class

Any subclass of TestCase responds to the message suite, which will build a
test suite that contains all the methods in the class whose names start with
the string “test”. To run the tests in the suite, send it the message run. For
example:

ExampleSetTest suite run −→ 5 run, 5 passed, 0 failed, 0 errors

Must I subclass TestCase?

In JUnit you can build a TestSuite from an arbitrary class containing test*
methods. In Smalltalk you can do the same but you will then have to

create a suite by hand and your class will have to implement all the essential
TestCase methods like assert:. We recommend that you not try to do this. The
framework is there: use it.

7.6 The SUnit framework

SUnit consists of four main classes: TestCase, TestSuite, TestResult, and
TestResource, as shown in Figure 7.3. The notion of a test resource was in-

troduced in SUnit 3.1 to represent a resource that is expensive to set-up but
which can be used by a whole series of tests. A TestResource specifies a setUp
method that is executed just once before a suite of tests; this is in distinction
to the TestCase»setUp method, which is executed before each test.

run
resources
addTest:

TestSuite

setUp
tearDown
assert:
deny:
should:raise:
shouldnt:raise:
selector:
run
resources

TestCase isAvailable
isUnavailable
setUp
tearDown

TestResource

passedCount
failuresCount
errorCount
runCount
tests

TestResulttests

Figure 7.3: The four classes representing the core of SUnit

TestCase

TestCase is an abstract class that is designed to be subclassed; each of its
subclasses represents a group of tests that share a common context (that is, a
test suite). Each test is run by creating a new instance of a subclass of TestCase,
running setUp, running the test method itself, and then running tearDown.

The context is specified by instance variables of the subclass and by the
specialization of the method setUp, which initializes those instance variables.
Subclasses of TestCase can also override method tearDown, which is invoked
after the execution of each test, and can be used to release any objects allocated
during setUp.

TestSuite

Instances of the class TestSuite contain a collection of test cases. An instance of
TestSuite contains tests, and other test suites. That is, a test suite contains sub-
instances of TestCase and TestSuite. Both individual TestCases and TestSuites
understand the same protocol, so they can be treated in the same way; for
example, both can be run. This is in fact an application of the composite
pattern in which TestSuite is the composite and the TestCases are the leaves —
see Design Patterns for more information on this pattern2.

2Erich Gamma et al., Design Patterns: Elements of Reusable Object-Oriented Software. Reading,
Mass.: Addison Wesley, 1995, ISBN 0–201–63361–2–(3).

TestResult

The class TestResult represents the results of a TestSuite execution. It records
the number of tests passed, the number of tests failed, and the number of
errors signalled.

TestResource

One of the important features of a suite of tests is that they should be inde-
pendent of each other: the failure of one test should not cause an avalanche of
failures of other tests that depend upon it, nor should the order in which the
tests are run matter. Performing setUp before each test and tearDown afterwards
helps to reinforce this independence.

However, there are occasions where setting up the necessary context is
just too time-consuming for it to be practical to do once before the execution
of each test. Moreover, if it is known that the test cases do not disrupt the
resources used by the tests, then it is wasteful to set them up afresh for each
test; it is sufficient to set them up once for each suite of tests. Suppose, for
example, that a suite of tests needs to query a database, or do some analysis
on some compiled code. In such cases, it may make sense to set up the
database and open a connection to it, or to compile some source code, before
any of the tests start to run.

Where should we cache these resources, so that they can be shared by
a suite of tests? The instance variables of a particular TestCase sub-instance
won’t do, because such an instance persists only for the duration of a single
test. A global variable would work, but using too many global variables
pollutes the name space, and the binding between the global and the tests
that depend on it will not be explicit. A better solution is to put the necessary
resources in a singleton object of some class. The class TestResource exists to be
subclassed by such resource classes. Each subclass of TestResource understands
the message current, which will answer a singleton instance of that subclass.
Methods setUp and tearDown should be overridden in the subclass to ensure
that the resource is initialized and finalized.

One thing remains: somehow, SUnit has to be told which resources are
associated with which test suite. A resource is associated with a particular
subclass of TestCase by overriding the class method resources. By default, the
resources of a TestSuite are the union of the resources of the TestCases that it
contains.

Here is an example. We define a subclass of TestResource called
MyTestResource and we associate it with MyTestCase by specializing the class
method resources to return an array of the test classes that it will use.

Class 7.8: An example of a TestResource subclass
TestResource subclass: #MyTestResource

instanceVariableNames: ''

MyTestCase class»resources
"associate the resource with this class of test cases"
↑{ MyTestResource }

7.7 Advanced features of SUnit

In addition to TestResource, the current version of SUnit contains assertion
description strings, logging support, and resumable test failures.

Assertion description strings

The TestCase assertion protocol includes a number of methods that allow the
programmer to supply a description of the assertion. The description is a
String; if the test case fails, this string will be displayed by the test runner. Of
course, this string can be constructed dynamically.

| e |
e := 42.
self assert: e = 23

description: 'expected 23, got ', e printString

The relevant methods in TestCase are:

#assert:description:
#deny:description:
#should:description:
#shouldnt:description:

Logging support

The description strings described above may also be logged to a Stream such
as the Transcript, or a file stream. You can choose whether to log by overriding
TestCase»isLogging in your test class; you must also choose where to log by
overriding TestCase»failureLog to answer an appropriate stream.

Continuing after a failure

SUnit also allows us to specify whether or not a test should continue after a
failure. This is a really powerful feature that uses the exception mechanisms

offered by Smalltalk. To see what this can be used for, let’s look at an example.
Consider the following test expression:

aCollection do: [ :each | self assert: each even]

In this case, as soon as the test finds the first element of the collection that
isn’t even, the test stops. However, we would usually like to continue, and
see both how many elements, and which elements, aren’t even, and maybe
also log this information. You can do this as follows:

aCollection do:
[:each |
self

assert: each even
description: each printString , ' is not even'
resumable: true]

This will print out a message on your logging stream for each element that
fails. It doesn’t accumulate failures, i.e., if the assertion fails 10 times in your
test method, you’ll still only see one failure. All the other assertion methods
that we have seen are not resumable; assert: p description: s is equivalent to
assert: p description: s resumable: false.

7.8 The implementation of SUnit

The implementation of SUnit makes an interesting case study of a Smalltalk
framework. Let’s look at some key aspects of the implementation by following
the execution of a test.

Running one test

To execute one test, we evaluate the expression (aTestClass selector: aSymbol)
run.

The method TestCase»run creates an instance of TestResult that will accu-
mulate the results of the tests, then it sends itself the message run:. (See
Figure 7.4.)

Method 7.9: Running a test case
TestCase»run

| result |
result := TestResult new.
self run: result.
↑result

run:

setUp

performTest

tearDown

:TestCase :TestResult

run

runCase:

runCase

Figure 7.4: Running one test

The method TestCase»run: sends the message runCase: to the test result:

Method 7.10: Passing the test case to the test result
TestCase»run: aResult

aResult runCase: self

The method TestResult»runCase: sends the message runCase to an individual
test, to execute the test. TestResult»runCase deals with any exceptions that may
be raised during the execution of a test, runs a TestCase by sending it the
message runCase, and counts the errors, failures and passes.

Method 7.11: Catching test case errors and failures
TestResult»runCase: aTestCase

| testCasePassed |
testCasePassed := true.
[[aTestCase runCase]

on: self class failure
do:

[:signal |
failures add: aTestCase.
testCasePassed := false.
signal return: false]]

on: self class error
do:



The implementation of SUnit 161

[:signal |
errors add: aTestCase.
testCasePassed := false.
signal return: false].

testCasePassed ifTrue: [passed add: aTestCase]

The method TestCase»runCase sends the messages setUp and tearDown as
shown below.

Method 7.12: Test case template method
TestCase»runCase

[self setUp.
self performTest] ensure: [self tearDown]

Running a TestSuite

To run more than one test, we send the message run to a TestSuite that contains
the relevant tests. TestCase class provides some functionality to build a test
suite from its methods. The expression MyTestCase buildSuiteFromSelectors re-
turns a suite containing all the tests defined in the MyTestCase class. The core
of this process is

Method 7.13: Auto-building the test suite
TestCase class»testSelectors
↑self selectors asSortedCollection asOrderedCollection select: [:each |

('test*' match: each) and: [each numArgs isZero]]

The method TestSuite»run creates an instance of TestResult, verifies that all
the resources are available, and then sends itself the message run:, which runs
all the tests in the suite. All the resources are then released.

Method 7.14: Running a test suite
TestSuite»run

| result |
result := TestResult new.
self resources do: [ :res |

res isAvailable ifFalse: [↑res signalInitializationError]].
[self run: result] ensure: [self resources do: [:each | each reset]].
↑result

Method 7.15: Passing the test result to the test suite
TestSuite»run: aResult

self tests do: [:each |
self changed: each.
each run: aResult].

The class TestResource and its subclasses keep track of their currently created
instances (one per class) that can be accessed and created using the class
method current. This instance is cleared when the tests have finished running
and the resources are reset.

The resource availability check makes it possible for the resource to be re-
created if needed, as shown in the class method TestResource class»isAvailable.
During the TestResource instance creation, it is initialized and the method
setUp is invoked.

Method 7.16: Test resource availability
TestResource class»isAvailable
↑self current notNil and: [self current isAvailable]

Method 7.17: Test resource creation
TestResource class»current

current isNil ifTrue: [current := self new].
↑current

Method 7.18: Test resource initialization
TestResource»initialize

super initialize.
self setUp

7.9 Some advice on testing

While the mechanics of testing are easy, writing good tests is not. Here is
some advice on how to design tests.

Feathers’ Rules for Unit tests. Michael Feathers, an agile process consultant
and author, writes:3

A test is not a unit test if:
• it talks to the database,
• it communicates across the network,
• it touches the file system,
• it can’t run at the same time as any of your other unit tests, or
• you have to do special things to your environment (such as

editing config files) to run it.

3See http://www.artima.com/weblogs/viewpost.jsp?thread=126923. 9 September 2005

http://www.artima.com/weblogs/viewpost.jsp?thread=126923

Tests that do these things aren’t bad. Often they are worth writing,
and they can be written in a unit test harness. However, it is
important to be able to separate them from true unit tests so that
we can keep a set of tests that we can run fast whenever we make
our changes.

Never get yourself into a situation where you don’t want to run your
unit test suite because it takes too long.

Unit Tests vs. Acceptance Tests. Unit tests capture one piece of functionality,
and as such make it easier to identify bugs in that functionality. As far
as possible try to have unit tests for each method that could possibly
fail, and group them per class. However, for certain deeply recursive
or complex setup situations, it is easier to write tests that represent
a scenario in the larger application; these are called acceptance tests
or functional tests. Tests that break Feathers’ rules may make good
acceptance tests. Group acceptance tests according to the functionality
that they test. For example, if you are writing a compiler, you might
write acceptance tests that make assertions about the code generated
for each possible source language statement. Such tests might exercise
many classes, and might take a long time to run because they touch the
file system. You can write them using SUnit, but you won’t want to run
them each time you make a small change, so they should be separated
from the true unit tests.

Black’s Rule of Testing. For every test in the system, you should be able to
identify some property for which the test increases your confidence. It’s
obvious that there should be no important property that you are not
testing. This rule states the less obvious fact that there should be no test
that does not add value to the system by increasing your confidence
that a useful property holds. For example, several tests of the same
property do no good. In fact, they do harm in two ways. First, they
make it harder to infer the behaviour of the class by reading the tests.
Second, because one bug in the code might then break many tests, they
make it harder to estimate how many bugs remain in the code. So, have
a property in mind when you write a test.

7.10 Chapter summary

This chapter explained why tests are an important investment in the future of
your code. We explained in a step-by-step fashion how to define a few tests for
the class Set. Then we gave an overview of the core of the SUnit framework by
presenting the classes TestCase, TestResult, TestSuite and TestResources. Finally

we looked deep inside SUnit by following the execution of a test and a test
suite.

• To maximize their potential, unit tests should be fast, repeatable, in-
dependent of any direct human interaction and cover a single unit of
functionality.

• Tests for a class called MyClass belong in a class classed MyClassTest,
which should be introduced as a subclass of TestCase.

• Initialize your test data in a setUp method.

• Each test method should start with the word “test”.

• Use the TestCase methods assert:, deny: and others to make assertions.

• Run tests using the SUnit test runner tool (in the tool bar).



Chapter 8

Basic Classes

Most of the magic of Smalltalk is not in the language but in the class libraries.
To program effectively with Smalltalk, you need to learn how the class li-
braries support the language and environment. The class libraries are entirely
written in Smalltalk and can easily be extended since a package may add new
functionality to a class even if it does not define this class.

Our goal here is not to present in tedious detail the whole of the Pharo
class library, but rather to point out the key classes and methods that you will
need to use or override to program effectively. In this chapter we cover the
basic classes that you will need for nearly every application: Object, Number
and its subclasses, Character, String, Symbol and Boolean.

8.1 Object

For all intents and purposes, Object is the root of the inheritance hierarchy.
Actually, in Pharo the true root of the hierarchy is ProtoObject, which is used
to define minimal entities that masquerade as objects, but we can ignore this
point for the time being.

Object can be found in the Kernel-Objects category. Astonishingly, there are
some 400 methods to be found here (including extensions). In other words,
every class that you define will automatically provide these 400 methods,
whether you know what they do or not. Note that some of the methods
should be removed and new versions of Pharo may remove some of the
superfluous methods.

The class comment for the Object states:

Object is the root class for almost all of the other classes in the class
hierarchy. The exceptions are ProtoObject (the superclass of Object) and
its subclasses. Class Object provides default behaviour common to all
normal objects, such as access, copying, comparison, error handling,
message sending, and reflection. Also utility messages that all objects
should respond to are defined here. Object has no instance variables,
nor should any be added. This is due to several classes of objects that
inherit from Object that have special implementations (SmallInteger and
UndefinedObject for example) or the VM knows about and depends on
the structure and layout of certain standard classes.

If we begin to browse the method categories on the instance side of Object
we start to see some of the key behaviour it provides.

Printing

Every object in Smalltalk can return a printed form of itself. You can select
any expression in a workspace and select the print it menu: this executes the
expression and asks the returned object to print itself. In fact this sends the
message printString to the returned object. The method printString, which is a
template method, at its core sends the message printOn: to its receiver. The
message printOn: is a hook that can be specialized.

Object»printOn: is very likely one of the methods that you will most fre-
quently override. This method takes as its argument a Stream on which a
String representation of the object will be written. The default implementation
simply writes the class name preceded by “a” or “an”. Object»printString returns
the String that is written.

For example, the class Browser does not redefine the method printOn: and
sending the message printString to an instance executes the methods defined
in Object.

Browser new printString −→ 'a Browser'

The class Color shows an example of printOn: specialization. It prints the
name of the class followed the name of the class method used to generate
that color.

Method 8.1: printOn: redefinition.
Color»printOn: aStream

| name |
(name := self name) ifNotNil:

[ ↑ aStream
nextPutAll: 'Color ';
nextPutAll: name ].

self storeOn: aStream

Color red printString −→ 'Color red'

Note that the message printOn: is not the same as storeOn:. The message
storeOn: puts on its argument stream an expression that can be used to recreate
the receiver. This expression is evaluated when the stream is read using the
message readFrom:. printOn: just returns a textual version of the receiver. Of
course, it may happen that this textual representation may represent the
receiver as a self-evaluating expression.

A word about representation and self-evaluating representation. In func-
tional programming, expressions return values when executed. In Smalltalk,
messages (expressions) return objects (values). Some objects have the nice
properties that their value is themselves. For example, the value of the object
true is itself i.e., the object true. We call such objects self-evaluating objects. You
can see a printed version of an object value when you print the object in a
workspace. Here are some examples of such self-evaluating expressions.

true −→ true
3@4 −→ 3@4
$a −→ $a
#(1 2 3) −→ #(1 2 3)
Color red −→ Color red

Note that some objects such as arrays are self-evaluating or not depend-
ing on the objects they contain. For example, an array of booleans is self-
evaluating whereas an array of persons is not. The following example shows
that a dynamic array is self-evaluating only if its elements are:

{10@10 . 100@100} −→ {10@10 . 100@100}
{Browser new . 100@100} −→ an Array(a Browser 100@100)

Remember that literal arrays can only contain literals. Hence the following
array does not contain two points but rather six literal elements.

#(10@10 100@100) −→ #(10 #@ 10 100 #@ 100)

Lots of printOn: method specializations implement self-evaluating behavior.
The implementations of Point»printOn: and Interval»printOn: are self-evaluating.

Method 8.2: Self-evaluation of Point

Point»printOn: aStream
"The receiver prints on aStream in terms of infix notation."
x printOn: aStream.
aStream nextPut: $@.
y printOn: aStream

Method 8.3: Self-evaluation of Interval

Interval»printOn: aStream
aStream nextPut: $(;

print: start;
nextPutAll: ' to: ';
print: stop.

step ∼= 1 ifTrue: [aStream nextPutAll: ' by: '; print: step].
aStream nextPut: $)

1 to: 10 −→ (1 to: 10) "intervals are self−evaluating"

Identity and equality

In Smalltalk, the message = tests object equality (i.e., whether two objects
represent the same value) whereas == tests object identity (i.e., whether two
expressions represent the same object).

The default implementation of object equality is to test for object identity:

Method 8.4: Object equality
Object»= anObject

"Answer whether the receiver and the argument represent the same object.
If = is redefined in any subclass, consider also redefining the message hash."
↑ self == anObject

This is a method that you will frequently want to override. Consider the
case of Complex numbers:

(1 + 2 i) = (1 + 2 i) −→ true "same value"
(1 + 2 i) == (1 + 2 i) −→ false "but different objects"

This works because Complex overrides = as follows:

Method 8.5: Equality for complex numbers
Complex»= anObject

anObject isComplex
ifTrue: [↑ (real = anObject real) & (imaginary = anObject imaginary)]
ifFalse: [↑ anObject adaptToComplex: self andSend: #=]

The default implementation of Object»∼= simply negates Object»=, and
should not normally need to be changed.

(1 + 2 i) ∼= (1 + 4 i) −→ true

If you override =, you should consider overriding hash. If instances of
your class are ever used as keys in a Dictionary, then you should make sure
that instances that are considered to be equal have the same hash value:

Method 8.6: Hash must be reimplemented for complex numbers
Complex»hash

"Hash is reimplemented because = is implemented."
↑ real hash bitXor: imaginary hash.

Although you should override = and hash together, you should never
override ==. (The semantics of object identity is the same for all classes.) == is
a primitive method of ProtoObject.

Note that Pharo has some strange behaviour compared to other Smalltalks:
for example a symbol and a string can be equal. (We consider this be a bug,
not a feature.)

#'lulu' = 'lulu' −→ true
'lulu' = #'lulu' −→ true

Class membership

Several methods allow you to query the class of an object.

class. You can ask any object about its class using the message class.

1 class −→ SmallInteger

Conversely, you can ask if an object is an instance of a specific class:

1 isMemberOf: SmallInteger −→ true "must be precisely this class"
1 isMemberOf: Integer −→ false
1 isMemberOf: Number −→ false
1 isMemberOf: Object −→ false

Since Smalltalk is written in itself, you can really navigate through its
structure using the right combination of superclass and class messages (see
Chapter 13).

isKindOf: Object»isKindOf: answers whether the receiver’s class is either the
same as, or a subclass of the argument class.

1 isKindOf: SmallInteger −→ true
1 isKindOf: Integer −→ true
1 isKindOf: Number −→ true
1 isKindOf: Object −→ true
1 isKindOf: String −→ false

1/3 isKindOf: Number −→ true
1/3 isKindOf: Integer −→ false

1/3 which is a Fraction is a kind of Number, since the class Number is a super-
class of the class Fraction, but 1/3 is not a Integer.

respondsTo: Object»respondsTo: answers whether the receiver understands
the message selector given as an argument.

1 respondsTo: #, −→ false

Normally it is a bad idea to query an object for its class, or to ask it which
messages it understands. Instead of making decisions based on the class of
object, you should simply send a message to the object and let it decide (i.e.,
on the basis of its class) how it should behave.

Copying

Copying objects introduces some subtle issues. Since instance variables are
accessed by reference, a shallow copy of an object would share its references to
instance variables with the original object:

a1 := { { 'harry' } }.
a1 −→ #(#('harry'))
a2 := a1 shallowCopy.
a2 −→ #(#('harry'))
(a1 at: 1) at: 1 put: 'sally'.
a1 −→ #(#('sally'))
a2 −→ #(#('sally')) "the subarray is shared"

Object»shallowCopy is a primitive method that creates a shallow copy of an
object. Since a2 is only a shallow copy of a1, the two arrays share a reference
to the nested Array that they contain.

Object»shallowCopy is the “public interface” to Object»copy and should be
overridden if instances are unique. This is the case, for example, with the
classes Boolean, Character, SmallInteger, Symbol and UndefinedObject.

Object»copyTwoLevel does the obvious thing when a simple shallow copy
does not suffice:

a1 := { { 'harry' } } .
a2 := a1 copyTwoLevel.
(a1 at: 1) at: 1 put: 'sally'.
a1 −→ #(#('sally'))
a2 −→ #(#('harry')) "fully independent state"

Object»deepCopy makes an arbitrarily deep copy of an object.

a1 := { { { 'harry' } } } .
a2 := a1 deepCopy.

(a1 at: 1) at: 1 put: 'sally'.
a1 −→ #(#('sally'))
a2 −→ #(#(#('harry')))

The problem with deepCopy is that it will not terminate when applied to a
mutually recursive structure:

a1 := { 'harry' }.
a2 := { a1 }.
a1 at: 1 put: a2.
a1 deepCopy −→ ... does not terminate!

Although it is possible to override deepCopy to do the right thing, Object»
copy offers a better solution:

Method 8.7: Copying objects as a template method
Object»copy

"Answer another instance just like the receiver.
Subclasses typically override postCopy;
they typically do not override shallowCopy."
↑self shallowCopy postCopy

You should override postCopy to copy any instance variables that should
not be shared. postCopy should always do a super postCopy.

Debugging

The most important method here is halt. In order to set a breakpoint in a
method, simply insert the message send self halt at some point in the body of
the method. When this message is sent, execution will be interrupted and a
debugger will open to this point in your program. (See Chapter 6 for more
details about the debugger.)

The next most important message is assert:, which takes a block as its
argument. If the block returns true, execution continues. Otherwise an
AssertionFailure exception will be raised. If this exception is not otherwise
caught, the debugger will open to this point in the execution. assert: is espe-
cially useful to support design by contract. The most typical usage is to check
non-trivial pre-conditions to public methods of objects. Stack»pop could easily
have been implemented as follows:

Method 8.8: Checking a pre-condition
Stack»pop

"Return the first element and remove it from the stack."
self assert: [ self isEmpty not ].
↑self linkedList removeFirst element

Do not confuse Object»assert: with TestCase»assert:, which occurs in the
SUnit testing framework (see Chapter 7). While the former expects a block
as its argument1, the latter expects a Boolean. Although both are useful for
debugging, they each serve a very different intent.

Error handling

This protocol contains several methods useful for signaling run-time errors.

Sending self deprecated: anExplanationString signals that the current method
should no longer be used, if deprecation has been turned on in the debug
protocol of the preference browser. The String argument should offer an
alternative.

1 doIfNotNil: [ :arg | arg printString, ' is not nil' ]
−→ SmallInteger(Object)»doIfNotNil: has been deprecated. use ifNotNilDo:

doesNotUnderstand: is sent whenever message lookup fails. The default
implementation, i.e., Object»doesNotUnderstand: will trigger the debugger at
this point. It may be useful to override doesNotUnderstand: to provide some
other behaviour.

Object»error and Object»error: are generic methods that can be used to raise
exceptions. (Generally it is better to raise your own custom exceptions, so
you can distinguish errors arising from your code from those coming from
kernel classes.)

Abstract methods in Smalltalk are implemented by convention with the
body self subclassResponsibility. Should an abstract class be instantiated by acci-
dent, then calls to abstract methods will result in Object»subclassResponsibility
being evaluated.

Method 8.9: Signaling that a method is abstract
Object»subclassResponsibility

"This message sets up a framework for the behavior of the class' subclasses.
Announce that the subclass should have implemented this message."
self error: 'My subclass should have overridden ', thisContext sender selector

printString

Magnitude, Number and Boolean are classical examples of abstract classes
that we shall see shortly in this chapter.

Number new + 1 −→ Error: My subclass should have overridden #+

self shouldNotImplement is sent by convention to signal that an inherited
method is not appropriate for this subclass. This is generally a sign that

1Actually, it will take any argument that understands value, including a Boolean.

something is not quite right with the design of the class hierarchy. Due to the
limitations of single inheritance, however, sometimes it is very hard to avoid
such workarounds.

A typical example is Collection»remove: which is inherited by Dictionary but
flagged as not implemented. (A Dictionary provides removeKey: instead.)

Testing

The testing methods have nothing to do with SUnit testing! A testing method
is one that lets you ask a question about the state of the receiver and returns a
Boolean.

Numerous testing methods are provided by Object. We have already
seen isComplex. Others include isArray, isBoolean, isBlock, isCollection and so on.
Generally such methods are to be avoided since querying an object for its
class is a form of violation of encapsulation. Instead of testing an object for
its class, one should simply send a request and let the object decide how to
handle it.

Nevertheless some of these testing methods are undeniably useful. The
most useful are probably ProtoObject»isNil and Object»notNil (though the Null
Object2 design pattern can obviate the need for even these methods).

Initialize release

A final key method that occurs not in Object but in ProtoObject is initialize.

Method 8.10: initialize as an empty hook method
ProtoObject»initialize

"Subclasses should redefine this method to perform initializations on instance creation"

The reason this is important is that in Pharo, the default new method
defined for every class in the system will send initialize to newly created
instances.

Method 8.11: new as a class-side template method
Behavior»new

"Answer a new initialized instance of the receiver (which is a class) with no indexable
variables. Fail if the class is indexable."
↑ self basicNew initialize

This means that simply by overriding the initialize hook method, new
instances of your class will automatically be initialized. The initialize method

2Bobby Woolf, Null Object. In Robert Martin, Dirk Riehle and Frank Buschmann, editors,
Pattern Languages of Program Design 3. Addison Wesley, 1998.

should normally perform a super initialize to establish the class invariant for
any inherited instance variables. (Note that this is not the standard behaviour
of other Smalltalks.)

8.2 Numbers

Remarkably, numbers in Smalltalk are not primitive data values but true
objects. Of course numbers are implemented efficiently in the virtual machine,
but the Number hierarchy is as perfectly accessible and extensible as any other
portion of the Smalltalk class hierarchy.

Number

Magnitude

Float Fraction Integer

SmallInteger LargePositiveInteger

LargeNegativeInteger

Object

Figure 8.1: The Number Hierarchy

Numbers are found in the Kernel-Numbers category. The abstract root of
this hierarchy is Magnitude, which represents all kinds of classes supporting
comparision operators. Number adds various arithmetic and other operators
as mostly abstract methods. Float and Fraction represent, respectively, floating
point numbers and fractional values. Integer is also abstract, thus distinguish-
ing between subclasses SmallInteger, LargePositiveInteger and LargeNegativeInteger.
For the most part users do not need to be aware of the difference between the
three Integer classes, as values are automatically converted as needed.

Magnitude

Magnitude is the parent not only of the Number classes, but also of other classes
supporting comparison operations, such as Character, Duration and Timespan.
(Complex numbers are not comparable, and so do not inherit from Number.)

Methods < and = are abstract. The remaining operators are generically
defined. For example:

Method 8.12: Abstract comparison methods
Magnitude» < aMagnitude

"Answer whether the receiver is less than the argument."
↑self subclassResponsibility

Magnitude» > aMagnitude
"Answer whether the receiver is greater than the argument."
↑aMagnitude < self

Number

Similarly, Number defines +, −, * and / to be abstract, but all other arithmetic
operators are generically defined.

All Number objects support various converting operators, such as asFloat
and asInteger. There are also numerous shortcut constructor methods, such as i,
which converts a Number to an instance of Complex with a zero real component,
and others which generate Durations, such as hour, day and week.

Numbers directly support common math functions such as sin, log, raiseTo:,
squared, sqrt and so on.

Number»printOn: is implemented in terms of the abstract method Number»
printOn:base:. (The default base is 10.)

Testing methods include even, odd, positive and negative. Unsurprisingly
Number overrides isNumber. More interesting, isInfinite is defined to return false.

Truncation methods include floor, ceiling, integerPart, fractionPart and so on.

1 + 2.5 −→ 3.5 "Addition of two numbers"
3.4 * 5 −→ 17.0 "Multiplication of two numbers"
8 / 2 −→ 4 "Division of two numbers"
10 − 8.3 −→ 1.7 "Subtraction of two numbers"
12 = 11 −→ false "Equality between two numbers"
12 ∼= 11 −→ true "Test if two numbers are different"
12 > 9 −→ true "Greater than"
12 >= 10 −→ true "Greater or equal than"
12 < 10 −→ false "Smaller than"
100@10 −→ 100@10 "Point creation"

The following example works surprisingly well in Smalltalk:

1000 factorial / 999 factorial −→ 1000

Note that 1000 factorial is really calculated which in many other languages
can be quite difficult to compute. This is an excellent example of automatic
coercion and exact handling of a number.

Try to display the result of 1000 factorial. It takes more time to display it than to
calculate it!

Float

Float implements the abstract Number methods for floating point numbers.

More interestingly, Float class (i.e., the class-side of Float) provides methods
to return the following constants: e, infinity, nan and pi.

Float pi −→ 3.141592653589793
Float infinity −→ Infinity
Float infinity isInfinite −→ true

Fraction

Fractions are represented by instance variables for the numerator and denomi-
nator, which should be Integers. Fractions are normally created by Integer divi-
sion (rather than using the constructor method Fraction»numerator:denominator:):

6/8 −→ (3/4)
(6/8) class −→ Fraction

Multiplying a Fraction by an Integer or another Fraction may yield an Integer:

6/8 * 4 −→ 3

Integer

Integer is the abstract parent of three concrete integer implementations. In
addition to providing concrete implementations of many abstract Number
methods, it also adds a few methods specific to integers, such as factorial,
atRandom, isPrime, gcd: and many others.

SmallInteger is special in that its instances are represented compactly —
instead of being stored as a reference, a SmallInteger is represented directly
using the bits that would otherwise be used to hold a reference. The first bit
of an object reference indicates whether the object is a SmallInteger or not.

The class methods minVal and maxVal tell us the range of a SmallInteger:

SmallInteger maxVal = ((2 raisedTo: 30) − 1) −→ true
SmallInteger minVal = (2 raisedTo: 30) negated −→ true

When a SmallInteger goes out of this range, it is automatically converted to
a LargePositiveInteger or a LargeNegativeInteger, as needed:

(SmallInteger maxVal + 1) class −→ LargePositiveInteger
(SmallInteger minVal − 1) class −→ LargeNegativeInteger

Large integers are similarly converted back to small integers when appro-
priate.

As in most programming languages, integers can be useful for specifying
iterative behaviour. There is a dedicated method timesRepeat: for evaluating a
block repeatedly. We have already seen a similar example in Chapter 3:

n := 2.
3 timesRepeat: [ n := n*n ].
n −→ 256

8.3 Characters

Character is defined in the Collections-Strings category as a subclass of Magnitude.
Printable characters are represented in Pharo as $〈char〉. For example:

$a < $b −→ true

Non-printing characters can be generated by various class methods.
Character class»value: takes the Unicode (or ASCII) integer value as argument
and returns the corresponding character. The protocol accessing untypeable
characters contains a number of convenience constructor methods such as
backspace, cr, escape, euro, space, tab, and so on.

Character space = (Character value: Character space asciiValue) −→ true

The printOn: method is clever enough to know which of the three ways to
generate characters offers the most appropriate representation:

Character value: 1 −→ Character home
Character value: 2 −→ Character value: 2
Character value: 32 −→ Character space
Character value: 97 −→ $a

Various convenient testing methods are built in: isAlphaNumeric, isCharacter,
isDigit, isLowercase, isVowel, and so on.

To convert a Character to the string containing just that character, send
asString. In this case asString and printString yield different results:

$a asString −→ 'a'
$a −→ $a
$a printString −→ '$a'

Every ascii Character is a unique instance, stored in the class variable
CharacterTable:

(Character value: 97) == $a −→ true

Characters outside the range 0 to 255 are not unique, however:

Character characterTable size −→ 256
(Character value: 500) == (Character value: 500) −→ false

8.4 Strings

The String class is also defined in the category Collections-Strings. A String is an
indexed Collection that holds only Characters.

Collection

ByteString

String

Symbol

Object

Array Text

ArrayedCollection

SequenceableCollection

Figure 8.2: The String Hierarchy

In fact, String is abstract and Pharo Strings are actually instances of the
concrete class ByteString.

'hello world' class −→ ByteString

The other important subclass of String is Symbol. The key difference is that
there is only ever a single instance of Symbol with a given value. (This is
sometimes called “the unique instance property”). In contrast, two separately
constructed Strings that happen to contain the same sequence of characters
will often be different objects.

'hel','lo' == 'hello' −→ false

('hel','lo') asSymbol == #hello −→ true

Another important difference is that a String is mutable, whereas a Symbol is
immutable.

'hello' at: 2 put: $u; yourself −→ 'hullo'

#hello at: 2 put: $u −→ error

It is easy to forget that since strings are collections, they understand the
same messages that other collections do:

#hello indexOf: $o −→ 5

Although String does not inherit from Magnitude, it does support the usual
comparing methods, <, = and so on. In addition, String»match: is useful for some
basic glob-style pattern-matching:

'*or*' match: 'zorro' −→ true

Should you need more advanced support for regular expressions, have a
look at the Regex package by Vassili Bykov.

Strings support rather a large number of conversion methods. Many
of these are shortcut constructor methods for other classes, such as asDate,
asFileName and so on. There are also a number of useful methods for convert-
ing a string to another string, such as capitalized and translateToLowercase.

For more on strings and collections, see Chapter 9.

8.5 Booleans

The class Boolean offers a fascinating insight into how much of the Smalltalk
language has been pushed into the class library. Boolean is the abstract super-
class of the Singleton classes True and False.

Most of the behaviour of Booleans can be understood by considering the
method ifTrue:ifFalse:, which takes two Blocks as arguments.

Object

ifTrue:IfFalse:
not
&

Boolean

ifTrue:IfFalse:
not
&

True

ifTrue:IfFalse:
not
&

False

Figure 8.3: The Boolean Hierarchy

(4 factorial > 20) ifTrue: [ 'bigger' ] ifFalse: [ 'smaller' ] −→ 'bigger'

The method is abstract in Boolean. The implementations in its concrete
subclasses are both trivial:

Method 8.13: Implementations of ifTrue:ifFalse:

True»ifTrue: trueAlternativeBlock ifFalse: falseAlternativeBlock
↑trueAlternativeBlock value

False»ifTrue: trueAlternativeBlock ifFalse: falseAlternativeBlock
↑falseAlternativeBlock value

In fact, this is the essence of OOP: when a message is sent to an object, the
object itself determines which method will be used to respond. In this case
an instance of True simply evaluates the true alternative, while an instance
of False evaluates the false alternative. All the abstract Boolean methods are
implemented in this way for True and False. For example:

Method 8.14: Implementing negation
True»not

"Negation−−answer false since the receiver is true."
↑false

Booleans offer several useful convenience methods, such as ifTrue:, ifFalse:,
ifFalse:ifTrue. You also have the choice between eager and lazy conjunctions
and disjunctions.

(1>2) & (3<4) −→ false "must evaluate both sides"
(1>2) and: [ 3<4 ] −→ false "only evaluate receiver"

(1>2) and: [ (1/0) > 0 ] −→ false "argument block is never evaluated, so no
exception"

In the first example, both Boolean subexpressions are evaluated, since &
takes a Boolean argument. In the second and third examples, only the first is
evaluated, since and: expects a Block as its argument. The Block is evaluated
only if the first argument is true.

Try to imagine how and: and or: are implemented. Check the implementations
in Boolean, True and False.

8.6 Chapter summary

• If you override = then you should override hash as well.

• Override postCopy to correctly implement copying for your objects.

• Send self halt to set a breakpoint.

• Return self subclassResponsibility to make a method abstract.

• To give an object a String representation you should override printOn:.

• Override the hook method initialize to properly initialize instances.

• Number methods automatically convert between Floats, Fractions and
Integers.

• Fractions truly represent rational numbers rather than floats.

• Characters are unique instances.

• Strings are mutable; Symbols are not. Take care not to mutate string
literals, however!

• Symbols are unique; Strings are not.

• Strings and Symbols are Collections and therefore support the usual
Collection methods.

Collections

9.1 Introduction

The collection classes form a loosely-defined group of general-purpose sub-
classes of Collection and Stream. The group of classes that appears in the “Blue
Book”1 contains 17 subclasses of Collection and 9 subclasses of Stream, for a
total of 28 classes, and had already been redesigned several times before the
Smalltalk-80 system was released. This group of classes is often considered
to be a paradigmatic example of object-oriented design.

In Pharo, the abstract class Collection has 101 subclasses, and the abstract
class Stream has 50 subclasses, but many of these (like Bitmap, FileStream and
CompiledMethod) are special-purpose classes crafted for use in other parts of
the system or in applications, and hence not categorized as “Collections” by
the system organization. For the purposes of this chapter, we use the term
“Collection Hierarchy” to mean Collection and its 47 subclasses that are also
in the categories labelled Collections-*. We use the term “Stream Hierarchy”
to mean Stream and its 9 subclasses that are also in the Collections-Streams
categories. These 56 classes respond to 982 messages and define a total of
1609 methods!

In this chapter we focus mainly on the subset of collection classes shown
in Figure 9.1. Streams will be discussed separately in Chapter 10.

1Adele Goldberg and David Robson, Smalltalk 80: the Language and its Implementation. Reading,
Mass.: Addison Wesley, May 1983, ISBN 0–201–13688–0.

Collection

String

Symbol

Object

Set

Dictionary

Array Text

Bag

OrderedCollection

SortedCollection

LinkedList

ArrayedCollection

Interval

SequenceableCollection

PluggableSet

PluggableDictionary

IdentityDictionary

ByteString

Figure 9.1: Some of the key collection classes in Pharo.

9.2 The varieties of collections

To make good use of the collection classes, the reader needs at least a superfi-
cial knowledge of the wide variety of collections that they implement, and
their commonalities and differences.

Programming with collections rather than individual elements is an im-
portant way to raise the level of abstraction of a program. The Lisp function
map, which applies an argument function to every element of a list and
returns a new list containing the results is an early example of this style,
but Smalltalk-80 adopted collection-based programming as a central tenet.
Modern functional programming languages such as ML and Haskell have
followed Smalltalk’s lead.

Why is this a good idea? Suppose you have a data structure containing a
collection of student records, and wish to perform some action on all of the
students that meet some criterion. Programmers raised to use an imperative
language will immediately reach for a loop. But the Smalltalk programmer
will write:

students select: [ :each | each gpa < threshold ]

which evaluates to a new collection containing precisely those elements of
students for which the bracketed function returns true2. The Smalltalk code

2The expression in brackets can be thought of as a λ-expression defining an anonymous

Protocol Methods
accessing size, capacity, at: anIndex , at: anIndex put: anElement
testing isEmpty, includes: anElement , contains: aBlock ,

occurrencesOf: anElement
adding add: anElement , addAll: aCollection
removing remove: anElement, remove: anElement ifAbsent: aBlock ,

removeAll: aCollection
enumerating do: aBlock , collect: aBlock , select: aBlock , reject: aBlock ,

detect: aBlock , detect: aBlock ifNone: aNoneBlock ,
inject: aValue into: aBinaryBlock

converting asBag, asSet, asOrderedCollection, asSortedCollection,
asArray, asSortedCollection: aBlock

creation with: anElement , with:with:, with:with:with:,
with:with:with:with:, withAll: aCollection

Figure 9.2: Standard Collection protocols

has the simplicity and elegance of a domain-specific query language.

The message select: is understood by all collections in Smalltalk. There is
no need to find out if the student data structure is an array or a linked list: the
select: message is understood by both. Note that this is quite different from
using a loop, where one must know whether students is an array or a linked
list before the loop can be set up.

In Smalltalk, when one speaks of a collection without being more specific
about the kind of collection, one means an object that supports well-defined
protocols for testing membership and enumerating the elements. All collec-
tions understand the testing messages includes:, isEmpty and occurrencesOf:. All
collections understand the enumeration messages do:, select:, reject: (which is
the opposite of select:), collect: (which is like lisp’s map), detect:ifNone:, inject:into:
(which performs a left fold) and many more. It is the ubiquity of this protocol,
as well as its variety, that makes it so powerful.

Figure 9.2 summarizes the standard protocols supported by most of the
classes in the collection hierarchy. These methods are defined, redefined,
optimized or occasionally even forbidden by subclasses of Collection.

Beyond this basic uniformity, there are many different kinds of collection
either supporting different protocols, or providing different behaviour for the
same requests. Let us briefly survey some of the key differences:

• Sequenceable: Instances of all subclasses of SequenceableCollection start
from a first element and proceed in a well-defined order to a last ele-

function λx.x gpa < threshold.

ment. Instances of Set, Bag and Dictionary, on the other hand, are not
sequenceable.

• Sortable: A SortedCollection maintains its elements in sort order.

• Indexable: Most sequenceable collections are also indexable, that is, ele-
ments can be retrieved with at:. Array is the familiar indexable data struc-
ture with a fixed size; anArray at: n retrieves the nth element of anArray,
and anArray at: n put: v changes the nth element to v. LinkedLists and SkipList
s are sequenceable but not indexable, that is, they understand first and
last, but not at:.

• Keyed: Instances of Dictionary and its subclasses are accessed by keys
instead of indices.

• Mutable: Most collections are mutable, but Intervals and Symbols are not.
An Interval is an immutable collection representing a range of Integers.
For example, 5 to: 16 by: 2 is an interval that contains the elements 5, 7, 9,
11, 13 and 15. It is indexable with at:, but cannot be changed with at:put:.

• Growable: Instances of Interval and Array are always of a fixed size. Other
kinds of collections (sorted collections, ordered collections, and linked
lists) can grow after creation.

The class OrderedCollection is more general than Array; the size of an
OrderedCollection grows on demand, and it has methods for addFirst: and
addLast: as well as at: and at:put:.

• Accepts duplicates: A Set will filter out duplicates, but a Bag will not.
Dictionary, Set and Bag use the = method provided by the elements; the
Identity variants of these classes use the == method, which tests whether
the arguments are the same object, and the Pluggable variants use an
arbitrary equivalence relation supplied by the creator of the collection.

• Heterogeneous: Most collections will hold any kind of element. A String
, CharacterArray or Symbol, however, only holds Characters. An Array will
hold any mix of objects, but a ByteArray only holds Bytes, an IntegerArray
only holds Integers and a FloatArray only holds Floats. A LinkedList is con-
strained to hold elements that conform to the Link .accessing protocol.

9.3 Implementations of collections

These categorizations by functionality are not our only concern; we must also
consider how the collection classes are implemented. As shown in Figure 9.3,
five main implementation techniques are employed.

Figure 9.3: Some collection classes categorized by implementation technique.

1. Arrays store their elements in the (indexable) instance variables of the
collection object itself; as a consequence, arrays must be of a fixed size,
but can be created with a single memory allocation.

2. OrderedCollections and SortedCollections store their elements in an array
that is referenced by one of the instance variables of the collection.
Consequently, the internal array can be replaced with a larger one if the
collection grows beyond its storage capacity.

3. The various kinds of set and dictionary also reference a subsidiary array
for storage, but use the array as a hash table. Bags use a subsidiary
Dictionary, with the elements of the bag as keys and the number of
occurrences as values.

4. LinkedLists use a standard singly-linked representation.

5. Intervals are represented by three integers that record the two endpoints
and the step size.

In addition to these classes, there are also “weak” variants of Array, Set and of
the various kinds of dictionary. These collections hold onto their elements
weakly, i.e., in a way that does not prevent the elements from being garbage
collected. The Pharo virtual machine is aware of these classes and handles
them specially.

Readers interested in learning more about the Smalltalk collections are
referred to LaLonde and Pugh’s excellent book3.

3Wilf LaLonde and John Pugh, Inside Smalltalk: Volume 1. Prentice Hall, 1990, ISBN 0–13–
468414–1.



188 Collections

9.4 Examples of key classes

We present now the most common or important collection classes using
simple code examples. The main protocols of collections are: at:, at:put: — to
access an element, add:, remove: — to add or remove an element, size, isEmpty,
include: — to get some information about the collection, do:, collect:, select: —
to iterate over the collection. Each collection may implement or not such
protocols, and when they do, they interpret them to fit with their semantics.
We suggest you browse the classes themselves to identify specific and more
advanced protocols.

We will focus on the most common collection classes: OrderedCollection, Set,
SortedCollection, Dictionary, Interval, and Array.

Common creation protocol. There are several ways to create instances of
collections. The most generic ones use the methods new: and with:. new:
anInteger creates a collection of size anInteger whose elements will all be nil.
with: anObject creates a collection and adds anObject to the created collection.
Different collections will realize this behaviour differently.

You can create collections with initial elements using the methods with:,
with:with: etc. for up to six elements.

Array with: 1 −→ #(1)
Array with: 1 with: 2 −→ #(1 2)
Array with: 1 with: 2 with: 3 −→ #(1 2 3)
Array with: 1 with: 2 with: 3 with: 4 −→ #(1 2 3 4)
Array with: 1 with: 2 with: 3 with: 4 with: 5 −→ #(1 2 3 4 5)
Array with: 1 with: 2 with: 3 with: 4 with: 5 with: 6 −→ #(1 2 3 4 5 6)

You can also use addAll: to add all elements of one kind of collection to
another kind:

(1 to: 5) asOrderedCollection addAll: '678'; yourself −→ an OrderedCollection(1 2 3
4 5 $6 $7 $8)

Take care that addAll: also returns its argument, and not the receiver!

You can also create many collections with withAll: or newFrom:

Array withAll: #(7 3 1 3) −→ #(7 3 1 3)
OrderedCollection withAll: #(7 3 1 3) −→ an OrderedCollection(7 3 1 3)
SortedCollection withAll: #(7 3 1 3) −→ a SortedCollection(1 3 3 7)
Set withAll: #(7 3 1 3) −→ a Set(7 1 3)
Bag withAll: #(7 3 1 3) −→ a Bag(7 1 3 3)
Dictionary withAll: #(7 3 1 3) −→ a Dictionary(1−>7 2−>3 3−>1 4−>3 )

Array newFrom: #(7 3 1 3) −→ #(7 3 1 3)
OrderedCollection newFrom: #(7 3 1 3) −→ an OrderedCollection(7 3 1

3)
SortedCollection newFrom: #(7 3 1 3) −→ a SortedCollection(1 3 3 7)
Set newFrom: #(7 3 1 3) −→ a Set(7 1 3)
Bag newFrom: #(7 3 1 3) −→ a Bag(7 1 3 3)
Dictionary newFrom: {1 −> 7. 2 −> 3. 3 −> 1. 4 −> 3} −→ a Dictionary(1−>7 2−>3 3−

>1 4−>3 )

Note that these two methods are not identical. In particular, Dictionary class
»withAll: interprets its argument as a collection of values, whereas Dictionary
class»newFrom: expects a collection of associations.

Array

An Array is a fixed-sized collection of elements accessed by integer indices.
Contrary to the C convention, the first element of a Smalltalk array is at
position 1 and not 0. The main protocol to access array elements is the
method at: and at:put:. at: anInteger returns the element at index anInteger. at:
anInteger put: anObject puts anObject at index anInteger. Arrays are fixed-size
collections therefore we cannot add or remove elements at the end of an
array. The following code creates an array of size 5, puts values in the first 3
locations and returns the first element.

anArray := Array new: 5.
anArray at: 1 put: 4.
anArray at: 2 put: 3/2.
anArray at: 3 put: 'ssss'.
anArray at: 1 −→ 4

There are several ways to create instances of the class Array. We can use
new:, with:, and the constructs #( ) and { }.

Creation with new: new: anInteger creates an array of size anInteger. Array new:
5 creates an array of size 5.

Creation with with: with: methods allows one to specify the value of the
elements. The following code creates an array of three elements consisting of
the number 4, the fraction 3/2 and the string 'lulu'.

Array with: 4 with: 3/2 with: 'lulu' −→ {4 . (3/2) . 'lulu'}

Literal creation with #(). #() creates literal arrays with static (or “literal”)
elements that have to be known when the expression is compiled, and not
when it is executed. The following code creates an array of size 2 where the
first element is the (literal) number 1 and the second the (literal) string 'here'.

#(1 'here') size −→ 2

Now, if you evaluate #(1+2), you do not get an array with a single element
3 but instead you get the array #(1 #+ 2) i.e., with three elements: 1, the symbol
#+ and the number 2.

#(1+2) −→ #(1 #+ 2)

This occurs because the construct #() causes the compiler to interpret literally
the expressions contained in the array. The expression is scanned and the
resulting elements are fed to a new array. Literal arrays contain numbers, nil,
true, false, symbols and strings.

Dynamic creation with { }. Finally, you can create a dynamic array using
the construct {}. { a . b } is equivalent to Array with: a with: b. This means in
particular that the expressions enclosed by { and } are executed.

{ 1 + 2 } −→ #(3)
{(1/2) asFloat} at: 1 −→ 0.5
{10 atRandom . 1/3} at: 2 −→ (1/3)

Element Access. Elements of all sequenceable collections can be accessed
with at: and at:put:.

anArray := #(1 2 3 4 5 6) copy.
anArray at: 3 −→ 3
anArray at: 3 put: 33.
anArray at: 3 −→ 33

Be careful with code that modifies literal arrays! The compiler tries to allocate
space just once for literal arrays. Unless you copy the array, the second time
you evaluate the code your “literal” array may not have the value you expect.
(Without cloning, the second time around, the literal #(1 2 3 4 5 6) will actually
be #(1 2 33 4 5 6)!) Dynamic arrays do not have this problem.

OrderedCollection

OrderedCollection is one of the collections that can grow, and to which elements
can be added sequentially. It offers a variety of methods such as add:, addFirst:,
addLast:, and addAll:.

ordCol := OrderedCollection new.
ordCol add: 'Seaside'; add: 'SqueakSource'; addFirst: 'Monticello'.
ordCol −→ an OrderedCollection('Monticello' 'Seaside' 'SqueakSource')

Removing Elements. The method remove: anObject removes the first occur-
rence of an object from the collection. If the collection does not include such
an object, it raises an error.

ordCol add: 'Monticello'.
ordCol remove: 'Monticello'.
ordCol −→ an OrderedCollection('Seaside' 'SqueakSource' 'Monticello')

There is a variant of remove: named remove:ifAbsent: that allows one to
specify as second argument a block that is executed in case the element to be
removed is not in the collection.

res := ordCol remove: 'zork' ifAbsent: [33].
res −→ 33

Conversion. It is possible to get an OrderedCollection from an Array (or any
other collection) by sending the message asOrderedCollection:

#(1 2 3) asOrderedCollection −→ an OrderedCollection(1 2 3)
'hello' asOrderedCollection −→ an OrderedCollection($h $e $l $l $o)

Interval

The class Interval represents ranges of numbers. For example, the interval of
numbers from 1 to 100 is defined as follows:

Interval from: 1 to: 100 −→ (1 to: 100)

The printString of this interval reveals that the class Number provides us with a
convenience method called to: to generate intervals:

(Interval from: 1 to: 100) = (1 to: 100) −→ true

We can use Interval class»from:to:by: or Number»to:by: to specify the step be-
tween two numbers as follow:

(Interval from: 1 to: 100 by: 0.5) size −→ 199
(1 to: 100 by: 0.5) at: 198 −→ 99.5
(1/2 to: 54/7 by: 1/3) last −→ (15/2)

Dictionary

Dictionaries are important collections whose elements are accessed using
keys. Among the most commonly used messages of dictionary you will find
at:, at:put:, at:ifAbsent:, keys and values.

colors := Dictionary new.
colors at: #yellow put: Color yellow.
colors at: #blue put: Color blue.
colors at: #red put: Color red.
colors at: #yellow −→ Color yellow
colors keys −→ a Set(#blue #yellow #red)
colors values −→ {Color blue . Color yellow . Color red}

Dictionaries compare keys by equality. Two keys are considered to be the
same if they return true when compared using =. A common and difficult to
spot bug is to use as key an object whose = method has been redefined but not
its hash method. Both methods are used in the implementation of dictionary
and when comparing objects.

The class Dictionary clearly illustrates that the collection hierarchy is based
on subclassing and not subtyping. Even though Dictionary is a subclass of Set,
we would normally not want to use a Dictionary where a Set is expected. In its
implementation, however, a Dictionary can clearly be seen as consisting of a
set of associations (key value) created using the message −>. We can create a
Dictionary from a collection of associations, or we may convert a dictionary to
an array of associations.

colors := Dictionary newFrom: { #blue−>Color blue. #red−>Color red. #yellow−>Color
yellow }.

colors removeKey: #blue.
colors associations −→ {#yellow−>Color yellow . #red−>Color red}

IdentityDictionary. While a dictionary uses the result of the messages =
and hash to determine if two keys are the same, the class IdentityDictionary uses
the identity (message ==) of the key instead of its values, i.e., it considers two
keys to be equal only if they are the same object.

Often Symbols are used as keys, in which case it is natural to use an
IdentityDictionary, since a Symbol is guaranteed to be globally unique. If, on the
other hand, your keys are Strings, it is better to use a plain Dictionary, or you
may get into trouble:

a := 'foobar'.
b := a copy.
trouble := IdentityDictionary new.
trouble at: a put: 'a'; at: b put: 'b'.
trouble at: a −→ 'a'

trouble at: b −→ 'b'
trouble at: 'foobar' −→ 'a'

Since a and b are different objects, they are treated as different objects. Inter-
estingly, the literal 'foobar' is allocated just once, so is really the same object
as a. You don’t want your code to depend on behaviour like this! A plain
Dictionary would give the same value for any key equal to 'foobar'.

Use only globally unique objects (like Symbols or SmallIntegers) as keys for
a IdentityDictionary, and Strings (or other objects) as keys for a plain Dictionary.

Note that the global Smalltalk is an instance of SystemDictionary, a subclass
of IdentityDictionary, hence all its keys are Symbols (actually, ByteSymbols, which
contain only 8-bit characters).

Smalltalk keys collect: [ :each | each class ] −→ a Set(ByteSymbol)

Sending keys or values to a Dictionary results in a Set, which we look at next.

Set

The class Set is a collection which behaves as a mathematical set, i.e., as
a collection with no duplicate elements and without any order. In a Set
elements are added using the message add: and they cannot be accessed using
the message at:. Objects put in a set should implement the methods hash and
=.

s := Set new.
s add: 4/2; add: 4; add:2.
s size −→ 2

You can also create sets using Set class»newFrom: or the conversion message
Collection»asSet:

(Set newFrom: #( 1 2 3 1 4 )) = #(1 2 3 4 3 2 1) asSet −→ true

asSet offers us a convenient way to eliminate duplicates from a collection:

{ Color black. Color white. (Color red + Color blue + Color green) } asSet size −→ 2

Note that red + blue + green = white.

A Bag is much like a Set except that it does allow duplicates:

{ Color black. Color white. (Color red + Color blue + Color green) } asBag size −→ 3

The set operations union, intersection and membership test are implemented
by the Collection messages union:, intersection: and includes:. The receiver is first
converted to a Set, so these operations work for all kinds of collections!

(1 to: 6) union: (4 to: 10) −→ a Set(1 2 3 4 5 6 7 8 9 10)
'hello' intersection: 'there' −→ 'he'
#Smalltalk includes: $k −→ true

As we explain below, elements of a set are accessed using iterators (see
Section 9.5).

SortedCollection

In contrast to an OrderedCollection, a SortedCollection maintains its elements in
sort order. By default, a sorted collection uses the message <= to establish
sort order, so it can sort instances of subclasses of the abstract class Magnitude,
which defines the protocol of comparable objects (<, =, >, >=, between:and:...).
(See Chapter 8.)

You can create a SortedCollection by creating a new instance and adding
elements to it:

SortedCollection new add: 5; add: 2; add: 50; add: −10; yourself. −→ a
SortedCollection(−10 2 5 50)

More usually, though, one will send the conversion message asSortedCollection
to an existing collection:

#(5 2 50 −10) asSortedCollection −→ a SortedCollection(−10 2 5 50)

This example answers the following FAQ:

FAQ: How do you sort a collection?
ANSWER: Send the message asSortedCollection to it.

'hello' asSortedCollection −→ a SortedCollection($e $h $l $l $o)

How do you get a String back from this result? asString unfortunately
returns the printString representation, which is not what we want:

'hello' asSortedCollection asString −→ 'a SortedCollection($e $h $l $l $o)'

The correct answer is to either use String class»newFrom:, String class»withAll: or
Object»as::

'hello' asSortedCollection as: String −→ 'ehllo'
String newFrom: ('hello' asSortedCollection) −→ 'ehllo'
String withAll: ('hello' asSortedCollection) −→ 'ehllo'

It is possible to have different kinds of elements in a SortedCollection as
long as they are all comparable. For example we can mix different kinds of
numbers such as integers, floats and fractions:

{ 5. 2/−3. 5.21 } asSortedCollection −→ a SortedCollection((−2/3) 5 5.21)

Imagine that you want to sort objects that do not define the method <= or
that you would like to have a different sorting criterion. You can do this by
supplying a two argument block, called a sortblock, to the sorted collection.
For example, the class Color is not a Magnitude and it does not implement the
method <=, but we can specify a block stating that the colors should be sorted
according to their luminance (a measure of brightness).

col := SortedCollection sortBlock: [:c1 :c2 | c1 luminance <= c2 luminance].
col addAll: { Color red. Color yellow. Color white. Color black }.
col −→ a SortedCollection(Color black Color red Color yellow Color white)

String

A Smalltalk String represents a collection of Characters. It is sequenceable,
indexable, mutable and homogeneous, containing only Character instances.
Like Arrays, Strings have a dedicated syntax, and are normally created by
directly specifying a String literal within single quotes, but the usual collection
creation methods will work as well.

'Hello' −→ 'Hello'
String with: $A −→ 'A'
String with: $h with: $i with: $! −→ 'hi!'
String newFrom: #($h $e $l $l $o) −→ 'hello'

In actual fact, String is abstract. When we instantiate a String we actually
get either an 8-bit ByteString or a 32-bit WideString. To keep things simple, we
usually ignore the difference and just talk about instances of String.

Two instances of String can be concatenated with a comma.

s := 'no', ' ', 'worries'.
s −→ 'no worries'

Since a string is a mutable collection we can also change it using the
method at:put:.

s at: 4 put: $h; at: 5 put: $u.
s −→ 'no hurries'

Note that the comma method is defined by Collection, so it will work for
any kind of collection!

(1 to: 3) , '45' −→ #(1 2 3 $4 $5)

We can also modify an existing string using replaceAll:with: or
replaceFrom:to:with: as shown below. Note that the number of characters and
the interval should have the same size.

s replaceAll: $n with: $N.
s −→ 'No hurries'
s replaceFrom: 4 to: 5 with: 'wo'.
s −→ 'No worries'

In contrast to the methods described above, the method copyReplaceAll:
creates a new string. (Curiously, here the arguments are substrings rather
than individual characters, and their sizes do not have to match.)

s copyReplaceAll: 'rries' with: 'mbats' −→ 'No wombats'

A quick look at the implementation of these methods reveals that they are
defined not only for Strings, but for any kind of SequenceableCollection, so the
following also works:

(1 to: 6) copyReplaceAll: (3 to: 5) with: { 'three'. 'etc.' } −→ #(1 2 'three' 'etc.' 6)

String matching. It is possible to ask whether a pattern matches a string by
sending the match: message. The pattern can specify * to match an arbitrary
series of characters and # to match a single character. Note that match: is sent
to the pattern and not the string to be matched.

'Linux *' match: 'Linux mag' −→ true
'GNU/Linux #ag' match: 'GNU/Linux tag' −→ true

Another useful method is findString:.

'GNU/Linux mag' findString: 'Linux' −→ 5
'GNU/Linux mag' findString: 'linux' startingAt: 1 caseSensitive: false −→ 5

More advanced pattern matching facilities offering the capabilities of Perl
are also available in the Regex package.

Some tests on strings. The following examples illustrate the use of isEmpty,
includes: and anySatisfy: which are further messages defined not only on Strings
but more generally on collections.

'Hello' isEmpty −→ false
'Hello' includes: $a −→ false
'JOE' anySatisfy: [:c | c isLowercase] −→ false
'Joe' anySatisfy: [:c | c isLowercase] −→ true

String templating. There are three messages that are useful to manage
string templating: format:, expandMacros and expandMacrosWith:.

'{1} is {2}' format: {'Pharo' . 'cool'} −→ 'Pharo is cool'

The messages of the expandMacros family offer variable substitution,
using <n> for carriage return, <t> for tabulation, <1s>, <2s>, <3s> for arguments
(<1p>, <2p>, surrounds the string with single quotes), and <1?value1:value2> for
conditional.

'look−<t>−here' expandMacros −→ 'look− −here'
'<1s> is <2s>' expandMacrosWith: 'Pharo' with: 'cool' −→ 'Pharo is cool'
'<2s> is <1s>' expandMacrosWith: 'Pharo' with: 'cool' −→ 'cool is Pharo'
'<1p> or <1s>' expandMacrosWith: 'Pharo' with: 'cool' −→ '''Pharo'' or Pharo'
'<1?Quentin:Thibaut> plays' expandMacrosWith: true −→ 'Quentin plays'
'<1?Quentin:Thibaut> plays' expandMacrosWith: false −→ 'Thibaut plays'

Some other utility methods. The class String offers numerous other utilities
including the messages asLowercase, asUppercase and capitalized.

'XYZ' asLowercase −→ 'xyz'
'xyz' asUppercase −→ 'XYZ'
'hilaire' capitalized −→ 'Hilaire'
'1.54' asNumber −→ 1.54
'this sentence is without a doubt far too long' contractTo: 20 −→ 'this sent...too long'

Note that there is generally a difference between asking an object its string
representation by sending the message printString and converting it to a string
by sending the message asString. Here is an example of the difference.

#ASymbol printString −→ '#ASymbol'
#ASymbol asString −→ 'ASymbol'

A symbol is similar to a string but is guaranteed to be globally unique.
For this reason symbols are preferred to strings as keys for dictionaries, in
particular for instances of IdentityDictionary. See also Chapter 8 for more about
String and Symbol.

9.5 Collection iterators

In Smalltalk loops and conditionals are simply messages sent to collections
or other objects such as integers or blocks (see also Chapter 3). In addition to
low-level messages such as to:do: which evaluates a block with an argument
ranging from an initial to a final number, the Smalltalk collection hierarchy
offers various high-level iterators. Using such iterators will make your code
more robust and compact.

Iterating (do:)

The method do: is the basic collection iterator. It applies its argument (a block
taking a single argument) to each element of the receiver. The following
example prints all the strings contained in the receiver to the transcript.

#('bob' 'joe' 'toto') do: [:each | Transcript show: each; cr].

Variants. There are a lot of variants of do:, such as do:without:, doWithIndex: and
reverseDo:: For the indexed collections (Array, OrderedCollection, SortedCollection)
the method doWithIndex: also gives access to the current index. This method is
related to to:do: which is defined in class Number.

#('bob' 'joe' 'toto') doWithIndex: [:each :i | (each = 'joe') ifTrue: [ ↑ i ] ] −→ 2

For ordered collections, reverseDo: walks the collection in the reverse order.

The following code shows an interesting message: do:separatedBy: which
executes the second block only in between two elements.

res := ''.
#('bob' 'joe' 'toto') do: [:e | res := res, e ] separatedBy: [res := res, '.'].
res −→ 'bob.joe.toto'

Note that this code is not especially efficient since it creates intermediate
strings and it would be better to use a write stream to buffer the result (see
Chapter 10):

String streamContents: [:stream | #('bob' 'joe' 'toto') asStringOn: stream delimiter: '.' ]
−→ 'bob.joe.toto'

Dictionaries. When the message do: is sent to a dictionary, the elements
taken into account are the values, not the associations. The proper methods
to use are keysDo:, valuesDo:, and associationsDo:, which iterate respectively on
keys, values or associations.

colors := Dictionary newFrom: { #yellow −> Color yellow. #blue −> Color blue. #red −>
Color red }.

colors keysDo: [:key | Transcript show: key; cr]. "displays the keys"
colors valuesDo: [:value | Transcript show: value;cr]. "displays the values"
colors associationsDo: [:value | Transcript show: value;cr]. "displays the associations"

Collecting results (collect:)

If you want to process the elements of a collection and produce a new col-
lection as a result, rather than using do:, you are probably better off using

collect:, or one of the other iterator methods. Most of these can be found in the
enumerating protocol of Collection and its subclasses.

Imagine that we want a collection containing the doubles of the elements
in another collection. Using the method do: we must write the following:

double := OrderedCollection new.
#(1 2 3 4 5 6) do: [:e | double add: 2 * e].
double −→ an OrderedCollection(2 4 6 8 10 12)

The method collect: executes its argument block for each element and returns a
new collection containing the results. Using collect: instead, the code is much
simpler:

#(1 2 3 4 5 6) collect: [:e | 2 * e] −→ #(2 4 6 8 10 12)

The advantages of collect: over do: are even more dramatic in the following
example, where we take a collection of integers and generate as a result a
collection of absolute values of these integers:

aCol := #( 2 −3 4 −35 4 −11).
result := aCol species new: aCol size.
1 to: aCol size do: [ :each | result at: each put: (aCol at: each) abs].
result −→ #(2 3 4 35 4 11)

Contrast the above with the much simpler following expression:

#( 2 −3 4 −35 4 −11) collect: [:each | each abs ] −→ #(2 3 4 35 4 11)

A further advantage of the second solution is that it will also work for sets
and bags.

Generally you should avoid using do:, unless you want to send messages
to each of the elements of a collection.

Note that sending the message collect: returns the same kind of collection
as the receiver. For this reason the following code fails. (A String cannot hold
integer values.)

'abc' collect: [:ea | ea asciiValue ] "error!"

Instead we must first convert the string to an Array or an OrderedCollection:

'abc' asArray collect: [:ea | ea asciiValue ] −→ #(97 98 99)

Actually collect: is not guaranteed to return a collection of exactly the same
class as the receiver, but only the same “species”. In the case of an Interval, the
species is actually an Array!

(1 to: 5) collect: [ :ea | ea * 2 ] −→ #(2 4 6 8 10)

Selecting and rejecting elements

select: returns the elements of the receiver that satisfy a particular condition:

(2 to: 20) select: [:each | each isPrime] −→ #(2 3 5 7 11 13 17 19)

reject: does the opposite:

(2 to: 20) reject: [:each | each isPrime] −→ #(4 6 8 9 10 12 14 15 16 18 20)

Identifying an element with detect:

The method detect: returns the first element of the receiver that matches block
argument.

'through' detect: [:each | each isVowel] −→ $o

The method detect:ifNone: is a variant of the method detect:. Its second block
is evaluated when there is no element matching the block.

Smalltalk allClasses detect: [:each | '*cobol*' match: each asString] ifNone: [ nil ]
−→ nil

Accumulating results with inject:into:

Functional programming languages often provide a higher-order function
called fold or reduce to accumulate a result by applying some binary operator
iteratively over all elements of a collection. In Pharo this is done by Collection»
inject:into:.

The first argument is an initial value, and the second argument is a two-
argument block which is applied to the result this far, and each element in
turn.

A trivial application of inject:into: is to produce the sum of a collection of
numbers. Following Gauss, in Pharo we could write this expression to sum
the first 100 integers:

(1 to: 100) inject: 0 into: [:sum :each | sum + each ] −→ 5050

Another example is the following one-argument block which computes
factorials:

factorial := [:n | (1 to: n) inject: 1 into: [:product :each | product * each ] ].
factorial value: 10 −→ 3628800

Other messages

count: The message count: returns the number of elements satisfying a con-
dition. The condition is represented as a boolean block.

Smalltalk allClasses count: [:each | 'Collection*' match: each asString ] −→ 3

includes: The message includes: checks whether the argument is contained in
the collection.

colors := {Color white . Color yellow. Color red . Color blue . Color orange}.
colors includes: Color blue. −→ true

anySatisfy: The message anySatisfy: answers true if at least one element of the
collection satisfies the condition represented by the argument.

colors anySatisfy: [:c | c red > 0.5] −→ true

9.6 Some hints for using collections

A common mistake with add: The following error is one of the most fre-
quent Smalltalk mistakes.

collection := OrderedCollection new add: 1; add: 2.
collection −→ 2

Here the variable collection does not hold the newly created collection but
rather the last number added. This is because the method add: returns the
element added and not the receiver.

The following code yields the expected result:

collection := OrderedCollection new.
collection add: 1; add: 2.
collection −→ an OrderedCollection(1 2)

You can also use the message yourself to return the receiver of a cascade of
messages:

collection := OrderedCollection new add: 1; add: 2; yourself −→ an
OrderedCollection(1 2)

Removing an element of the collection you are iterating on. Another mis-
take you may make is to remove an element from a collection you are currently
iterating over. remove:

range := (2 to: 20) asOrderedCollection.
range do: [:aNumber | aNumber isPrime ifFalse: [ range remove: aNumber ] ].
range −→ an OrderedCollection(2 3 5 7 9 11 13 15 17 19)

This result is clearly incorrect since 9 and 15 should have been filtered out!

The solution is to copy the collection before going over it.

range := (2 to: 20) asOrderedCollection.
range copy do: [:aNumber | aNumber isPrime ifFalse: [ range remove: aNumber ] ].
range −→ an OrderedCollection(2 3 5 7 11 13 17 19)

Redefining both = and hash. A difficult error to spot is when you redefine
= but not hash. The symptoms are that you will lose elements that you put in
sets or other strange behaviour. One solution proposed by Kent Beck is to use
xor: to redefine hash. Suppose that we want two books to be considered equal
if their titles and authors are the same. Then we would redefine not only =
but also hash as follows:

Method 9.1: Redefining = and hash.
Book»= aBook

self class = aBook class ifFalse: [↑ false].
↑ title = aBook title and: [ authors = aBook authors]

Book»hash
↑ title hash xor: authors hash

Another nasty problem arises if you use a mutable object, i.e., an object
that can change its hash value over time, as an element of a Set or as a key to
a Dictionary. Don’t do this unless you love debugging!

9.7 Chapter summary

The Smalltalk collection hierarchy provides a common vocabulary for uni-
formly manipulating a variety of different kinds of collections.

• A key distinction is between SequenceableCollections, which maintain
their elements in a given order, Dictionary and its subclasses, which main-
tain key-to-value associations, and Sets and Bags, which are unordered.

• You can convert most collections to another kind of collection by send-
ing them the messages asArray, asOrderedCollection etc..

• To sort a collection, send it the message asSortedCollection.

• Literal Arrays are created with the special syntax #( ... ). Dynamic Arrays
are created with the syntax { ... }.

• A Dictionary compares keys by equality. It is most useful when keys are
instances of String. An IdentityDictionary instead uses object identity to
compare keys. It is more suitable when Symbols are used as keys, or
when mapping object references to values.

• Strings also understand the usual collection messages. In addition, a
String supports a simple form of pattern-matching. For more advanced
application, look instead at the RegEx package.

• The basic iteration message is do:. It is useful for imperative code, such
as modifying each element of a collection, or sending each element a
message.

• Instead of using do:, it is more common to use collect:, select:, reject:,
includes:, inject:into: and other higher-level messages to process collections
in a uniform way.

• Never remove an element from a collection you are iterating over. If
you must modify it, iterate over a copy instead.

• If you override =, remember to override hash as well!

Streams

Streams are used to iterate over sequences of elements such as sequenced
collections, files, and network streams. Streams may be either readable,
or writeable, or both. Reading or writing is always relative to the current
position in the stream. Streams can easily be converted to collections, and
vice versa.

10.1 Two sequences of elements

A good metaphor to understand a stream is the following: A stream can
be represented as two sequences of elements: a past element sequence and
a future element sequence. The stream is positioned between the two se-
quences. Understanding this model is important since all stream operations
in Smalltalk rely on it. For this reason, most of the Stream classes are sub-
classes of PositionableStream. Figure 10.1 presents a stream which contains five
characters. This stream is in its original position, i.e., there is no element in
the past. You can go back to this position using the message reset.

a b c d e

future element
sequence

past element
sequence

Figure 10.1: A stream positioned at its beginning.

Reading an element conceptually means removing the first element of
the future element sequence and putting it after the last element in the past
element sequence. After having read one element using the message next, the
state of your stream is that shown in Figure 10.2.

a b c d e

future element
sequence

past element
sequence

Figure 10.2: The same stream after the execution of the method next: the
character a is “in the past” whereas b, c, d and e are “in the future”.

Writing an element means replacing the first element of the future se-
quence by the new one and moving it to the past. Figure 10.3 shows the
state of the same stream after having written an x using the message nextPut:
anElement.

a x c d e

future element
sequence

past element
sequence

Figure 10.3: The same stream after having written an x.

10.2 Streams vs. collections

The collection protocol supports the storage, removal and enumeration of
the elements of a collection, but does not allow these operations to be inter-
mingled. For example, if the elements of an OrderedCollection are processed
by a do: method, it is not possible to add or remove elements from inside
the do: block. Nor does the collection protocol offer ways to iterate over
two collections at the same time, choosing which collection goes forward
and which does not. Procedures like these require that a traversal index or
position reference is maintained outside of the collection itself: this is exactly
the role of ReadStream, WriteStream and ReadWriteStream.

These three classes are defined to stream over some collection. For exam-
ple, the following snippet creates a stream on an interval, then it reads two
elements.

r := ReadStream on: (1 to: 1000).
r next. −→ 1
r next. −→ 2
r atEnd. −→ false

WriteStreams can write data to the collection:

w := WriteStream on: (String new: 5).
w nextPut: $a.
w nextPut: $b.
w contents. −→ 'ab'

It is also possible to create ReadWriteStreams that support both the reading
and writing protocols.

The main problem with WriteStream and ReadWriteStream is that they only
support arrays and strings in Pharo. This is currently being changed by the
development of a new library named Nile, but for now if you try to stream
over another kind of collection, you will get an error:

w := WriteStream on: (OrderedCollection new: 20).
w nextPut: 12. −→ raises an error

Streams are not only meant for collections, they can be used for files or
sockets too. The following example creates a file named test.txt, writes two
strings to it, separated by a carriage return, and closes the file.

StandardFileStream
fileNamed: 'test.txt'
do: [:str | str

nextPutAll: '123';
cr;
nextPutAll: 'abcd'].

The following sections present the protocols in more depth.

10.3 Streaming over collections

Streams are really useful when dealing with collections of elements. They can
be used for reading and writing elements in collections. We will now explore
the stream features for the collections.

Reading collections

This section presents features used for reading collections. Using a stream to
read a collection essentially provides you a pointer into the collection. That
pointer will move forward on reading and you can place it wherever you
want. The class ReadStream should be used to read elements from collections.

Methods next and next: are used to retrieve one or more elements from the
collection.

stream := ReadStream on: #(1 (a b c) false).
stream next. −→ 1
stream next. −→ #(#a #b #c)
stream next. −→ false

stream := ReadStream on: 'abcdef'.
stream next: 0. −→ ''
stream next: 1. −→ 'a'
stream next: 3. −→ 'bcd'
stream next: 2. −→ 'ef'

The message peek is used when you want to know what is the next element
in the stream without going forward.

stream := ReadStream on: '−143'.
negative := (stream peek = $−). "look at the first element without reading it"
negative. −→ true
negative ifTrue: [stream next]. "ignores the minus character"
number := stream upToEnd.
number. −→ '143'

This code sets the boolean variable negative according to the sign of the number
in the stream and number to its absolute value. The method upToEnd returns
everything from the current position to the end of the stream and sets the
stream to its end. This code can be simplified using peekFor:, which moves
forward if the following element equals the parameter and doesn’t move
otherwise.

stream := '−143' readStream.
(stream peekFor: $−) −→ true
stream upToEnd −→ '143'

peekFor: also returns a boolean indicating if the parameter equals the element.

You might have noticed a new way of constructing a stream in the above
example: one can simply send readStream to a sequenceable collection to get a
reading stream on that particular collection.

Positioning. There are methods to position the stream pointer. If you have
the index, you can go directly to it using position:. You can request the current
position using position. Please remember that a stream is not positioned on
an element, but between two elements. The index corresponding to the
beginning of the stream is 0.

You can obtain the state of the stream depicted in Figure 10.4 with the
following code:

stream := 'abcde' readStream.
stream position: 2.
stream peek −→ $c

a b c d e

future element
sequence

past element
sequence

Figure 10.4: A stream at position 2

To position the stream at the beginning or the end, you can use reset or
setToEnd. skip: and skipTo: are used to go forward to a location relative to the
current position: skip: accepts a number as argument and skips that number
of elements whereas skipTo: skips all elements in the stream until it finds an
element equal to its parameter. Note that it positions the stream after the
matched element.

stream := 'abcdef' readStream.
stream next. −→ $a "stream is now positioned just after the a"
stream skip: 3. "stream is now after the d"
stream position. −→ 4
stream skip: −2. "stream is after the b"
stream position. −→ 2
stream reset.
stream position. −→ 0
stream skipTo: $e. "stream is just after the e now"
stream next. −→ $f
stream contents. −→ 'abcdef'

As you can see, the letter e has been skipped.

The method contents always returns a copy of the entire stream.

Testing. Some methods allow you to test the state of the current stream:
atEnd returns true if and only if no more elements can be read whereas isEmpty
returns true if and only if there is no element at all in the collection.

Here is a possible implementation of an algorithm using atEnd that takes
two sorted collections as parameters and merges those collections into another
sorted collection:

stream1 := #(1 4 9 11 12 13) readStream.
stream2 := #(1 2 3 4 5 10 13 14 15) readStream.

"The variable result will contain the sorted collection."

result := OrderedCollection new.
[stream1 atEnd not & stream2 atEnd not]
whileTrue: [stream1 peek < stream2 peek
"Remove the smallest element from either stream and add it to the result."
ifTrue: [result add: stream1 next]
ifFalse: [result add: stream2 next]].

"One of the two streams might not be at its end. Copy whatever remains."
result
addAll: stream1 upToEnd;
addAll: stream2 upToEnd.

result. −→ an OrderedCollection(1 1 2 3 4 4 5 9 10 11 12 13 13 14 15)

Writing to collections

We have already seen how to read a collection by iterating over its elements us-
ing a ReadStream. We’ll now learn how to create collections using WriteStreams.

WriteStreams are useful for appending a lot of data to a collection at various
locations. They are often used to construct strings that are based on static and
dynamic parts as in this example:

stream := String new writeStream.
stream
nextPutAll: 'This Smalltalk image contains: ';
print: Smalltalk allClasses size;
nextPutAll: ' classes.';
cr;
nextPutAll: 'This is really a lot.'.

stream contents. −→ 'This Smalltalk image contains: 2322 classes.
This is really a lot.'

This technique is used in the different implementations of the method
printOn: for example. There is a simpler and more efficient way of creating
streams if you are only interested in the content of the stream:

string := String streamContents:
[:stream |

stream
print: #(1 2 3);
space;
nextPutAll: 'size';
space;
nextPut: $=;
space;
print: 3. ].

string. −→ '#(1 2 3) size = 3'

The method streamContents: creates a collection and a stream on that col-
lection for you. It then executes the block you gave passing the stream as a
parameter. When the block ends, streamContents: returns the content of the
collection.

The following WriteStream methods are especially useful in this context:

nextPut: adds the parameter to the stream;

nextPutAll: adds each element of the collection, passed as a parameter, to the
stream;

print: adds the textual representation of the parameter to the stream.

There are also methods useful for printing different kinds of characters to
the stream like space, tab and cr (carriage return). Another useful method is
ensureASpace which ensures that the last character in the stream is a space; if
the last character isn’t a space it adds one.

About Concatenation. Using nextPut: and nextPutAll: on a WriteStream is often
the best way to concatenate characters. Using the comma concatenation
operator (,) is far less efficient:

[| temp |
temp := String new.
(1 to: 100000)
do: [:i | temp := temp, i asString, ' ']] timeToRun −→ 115176 "(milliseconds)"

[| temp |
temp := WriteStream on: String new.
(1 to: 100000)
do: [:i | temp nextPutAll: i asString; space].

temp contents] timeToRun −→ 1262 "(milliseconds)"

The reason that using a stream can be much more efficient is that comma
creates a new string containing the concatenation of the receiver and the
argument, so it must copy both of them. When you repeatedly concatenate
onto the same receiver, it gets longer and longer each time, so that the number
of characters that must be copied goes up exponentially. This also creates a
lot of garbage, which must be collected. Using a stream instead of string con-
catenation is a well-known optimization. In fact, you can use streamContents:
(mentioned on page 211) to help you do this:

String streamContents: [ :tempStream |
(1 to: 100000)

do: [:i | tempStream nextPutAll: i asString; space]]

Reading and writing at the same time

It’s possible to use a stream to access a collection for reading and writing at
the same time. Imagine you want to create an History class which will manage
backward and forward buttons in a web browser. A history would react as in
figures from 10.5 to 10.11.

Figure 10.5: A new history is empty. Nothing is displayed in the web browser.

page1

Figure 10.6: The user opens to page 1.

page2page1

Figure 10.7: The user clicks on a link to page 2.

page2 page3page1

Figure 10.8: The user clicks on a link to page 3.

This behaviour can be implemented using a ReadWriteStream.

page2 page3page1

Figure 10.9: The user clicks on the back button. He is now viewing page 2
again.

page2 page3page1

Figure 10.10: The user clicks again the back button. Page 1 is now displayed.

page4page1

Figure 10.11: From page 1, the user clicks on a link to page 4. The history
forgets pages 2 and 3.

Object subclass: #History
instanceVariableNames: 'stream'
classVariableNames: ''
poolDictionaries: ''
category: 'PBE−Streams'

History>>initialize
super initialize.
stream := ReadWriteStream on: Array new.

Nothing really difficult here, we define a new class which contains a
stream. The stream is created during the initialize method.

We need methods to go backward and forward:

History>>goBackward
self canGoBackward ifFalse: [self error: 'Already on the first element'].
stream skip: −2.
↑ stream next.

History>>goForward

self canGoForward ifFalse: [self error: 'Already on the last element'].
↑ stream next

Until then, the code was pretty straightforward. Now, we have to deal
with the goTo: method which should be activated when the user clicks on a
link. A possible solution is:

History>>goTo: aPage
stream nextPut: aPage.

This version is incomplete however. This is because when the user clicks
on the link, there should be no more future pages to go to, i.e., the forward
button must be deactivated. To do this, the simplest solution is to write nil
just after to indicate the history end:

History>>goTo: anObject
stream nextPut: anObject.
stream nextPut: nil.
stream back.

Now, only methods canGoBackward and canGoForward have to be imple-
mented.

A stream is always positioned between two elements. To go backward,
there must be two pages before the current position: one page is the current
page, and the other one is the page we want to go to.

History>>canGoBackward
↑ stream position > 1

History>>canGoForward
↑ stream atEnd not and: [stream peek notNil]

Let us add a method to peek at the contents of the stream:

History>>contents
↑ stream contents

And the history works as advertised:

History new
goTo: #page1;
goTo: #page2;
goTo: #page3;
goBackward;
goBackward;
goTo: #page4;
contents −→ #(#page1 #page4 nil nil)

10.4 Using streams for file access

You have already seen how to stream over collections of elements. It’s also
possible to stream over files on your hard disk. Once created, a stream on
a file is really like a stream on a collection: you will be able to use the same
protocol to read, write or position the stream. The main difference appears
in the creation of the stream. There are several different ways to create file
streams, as we shall now see.

Creating file streams

To create file streams, you will have to use one of the following instance
creation methods offered by the class FileStream:

fileNamed: Open a file with the given name for reading and writing. If the
file already exists, its prior contents may be modified or replaced, but
the file will not be truncated on close. If the name has no directory part,
then the file will be created in the default directory.

newFileNamed: Create a new file with the given name, and answer a stream
opened for writing on that file. If the file already exists, ask the user
what to do.

forceNewFileNamed: Create a new file with the given name, and answer a
stream opened for writing on that file. If the file already exists, delete it
without asking before creating the new file.

oldFileNamed: Open an existing file with the given name for reading and
writing. If the file already exists, its prior contents may be modified or
replaced, but the file will not be truncated on close. If the name has no
directory part, then the file will be created in the default directory.

readOnlyFileNamed: Open an existing file with the given name for reading.

You have to remember that each time you open a stream on a file, you
have to close it too. This is done through the close method.

stream := FileStream forceNewFileNamed: 'test.txt'.
stream

nextPutAll: 'This text is written in a file named ';
print: stream localName.

stream close.

stream := FileStream readOnlyFileNamed: 'test.txt'.
stream contents. −→ 'This text is written in a file named ''test.txt'''
stream close.


